

# EYEDB Object Query Language

---

Version 2.8.0

January 2006

Copyright © 2001-2006 SYSRA

Published by SYSRA  
30, avenue Général Leclerc  
91330 Yerres - France

home page: <http://www.eyedb.org>

# Contents

1	Introduction . . . . .	5
2	Principles . . . . .	5
3	OQL vs. ODMG 3 OQL . . . . .	5
4	Language Concepts . . . . .	7
5	Language Syntax . . . . .	8
5.1	Terminal Atom Syntax . . . . .	8
5.2	Non Terminal Atom Production . . . . .	10
5.3	Keywords . . . . .	11
5.4	Comments . . . . .	11
5.5	Statements . . . . .	12
5.6	Expression Statements . . . . .	12
5.7	Atomic Literal Expressions . . . . .	14
5.8	Arithmetic Expressions . . . . .	14
5.9	Assignment Expressions . . . . .	18
5.10	Auto Increment & Decrement Expressions . . . . .	19
5.11	Comparison Expressions . . . . .	20
5.12	Logical Expressions . . . . .	24
5.13	Conditional Expression . . . . .	24
5.14	Expression Sequences . . . . .	25
5.15	Array Referencing . . . . .	25
5.16	Identifier Expressions . . . . .	28
5.17	Path Expressions . . . . .	32
5.18	Function Call . . . . .	33
5.19	Method Invocation . . . . .	34
5.20	Eval/Unval Operators . . . . .	37
5.21	Set Expressions . . . . .	38
5.22	Object Creation . . . . .	40
5.23	Object Deletion . . . . .	42
5.24	Collection Expressions . . . . .	42
5.25	Exception Expressions . . . . .	49
5.26	Function Definition Expressions . . . . .	49
5.27	Conversion Expressions . . . . .	51
5.28	Type Information Expressions . . . . .	54
5.29	Query Expressions . . . . .	55
5.30	Miscellenaous Expressions . . . . .	59
5.31	Selection Statements . . . . .	60
5.32	Iteration Statements . . . . .	61
5.33	Jump Statements . . . . .	63
5.34	Function Definition Statements . . . . .	64
6	Quick Reference Manual . . . . .	68
6.1	Builtin and Library Functions and Methods . . . . .	68
6.2	Special Variables . . . . .	71
6.3	The <code>eyedboql</code> Tool . . . . .	71
6.4	The Standard Library Package . . . . .	75
6.5	OQL Quick Reference Card . . . . .	89



# The Object Query Language

## 1 Introduction

In this chapter, we present the EYEDB Object Query Language which supports the EYEDB object model. It is based on the ODMG 3 Object Query Language OQL.

We first describe the design principles of the language in Section 2, then we present in Section 3 the main differences between EYEDB OQL and ODMG OQL. The language concepts are presented in Section 4. In Section 5, we introduced the language syntax. A quick reference manual of OQL is given in Section 6.

In this chapter, OQL denotes the EYEDB Object Query Language while the standard ODMG 3 Object Query Language will be denoted as ODMG OQL.

## 2 Principles

The principles of OQL are close to the principles of ODMG OQL introduced in the book *Object Database Standard, ODMG 3* by Rick Cattell and al.

Our design is based on the following principles and assumptions:

- OQL relies on the EYEDB Object Model.
- OQL is based on ODMG OQL close to SQL 92. Extensions to SQL 92 concern object-oriented notions, like complex objects, object identify, path expressions, polymorphism, operation invocation and late binding. Extensions to ODMG OQL concern function definitions, selection statements, iteration statement, assignment operator (see Section 3).
- OQL provides high level primitives to deal with sets of objects, structures, lists and arrays.
- OQL is a functional language where operators can freely be composed as long as the operands respect the type system.
- OQL is computationally complete.
- OQL can be invoked from within programming languages for which an EYEDB binding is defined (currently C++ and Java). Conversely, OQL can invoke operations programmed in this language.

## 3 OQL vs. ODMG 3 OQL

OQL implements all the ODMG OQL functionalities with a few exceptions concerning the **select** clause. In order to accept the whole DML (Data Manipulation Language) query part of SQL as a valid syntax for ODMG OQL, ad-hoc constructions have been added to ODMG OQL each time SQL introduces a syntax that cannot be considered in the category of true operators.

For instance, the following construct is a valid ODMG OQL construct:

```
select p.name, salary from Professors p
```

This construct is not currently a valid OQL construct. The alternate valid OQL form is:

```
select struct(name: p.name, salary: p.salary) from Professors p
```

In the same way, ODMG OQL accepts SQL forms of the aggregate operators `min`, `max`, `count`, `sum` and `avg`, for instance:

```
select count(*) from Persons
select max(e.salary) from Employees e
```

These constructs are not currently valid OQL constructs. The alternate valid OQL forms are:

```
count(select p from Persons p)
max(select e.salary from Employees e)
```

In the same way, the `select *` clause is not currently implemented in OQL, neither the implicit `select` clause (i.e. without explicit identifier). For instance, the following constructs are not OQL valid constructs, although there are valid ODMG OQL constructs:

```
select * from Person
select name from Person
```

There is no alternate OQL valid form for the first construct. The alternate OQL valid forms for the second construct is:

```
select Person.name
select p.name from Person p
select p.name from Person as p
select p.name from p in Person
```

On the other hand, OQL provides a few extensions to ODMG OQL:

- assignment operators,
- four regular expression operators,
- selection statements, `if/else`,
- iteration statements, `while`, `do/while`, two forms of `for`,
- function definition statements, `function`,
- the `eval` and `unval` operators,
- identifier operators, `isset`, `unset`, `refof`, `valof`, `&`, `*`, `push`, `pop`,
- exception management operators, `throw`,
- type information operators, `classof`, `typeof`
- miscellaneous operators, `structof`, `bodyof`
- builtin and library functions.

For instance, the following constructs are valid OQL constructs:

```
for (x in l)
{
  if (classof x != "Person")
    throw "Person type expected";
  if (x->name ~ "^john")
  {
    ok := true;
    break;
  }
}
```

```
function fib(n) {
  if (n < 2)
    return n;
  return fib(n-1) + fib(n-2);
}
```

```
for (n := 0, v := 0; n < 15; n++)
  v += fib(n);
```

```
function swap(x, y) {
  v := *x;
  *x := *y;
  *y := v;
}
```

```

i := "ii"; j := "jj";
swap(&i, &j);

function get_from(classname, attrname) {
    return eval "select x." + attrname + " from " + classname + " x";
}

names := get_from("Person", "name");

```

These extensions make OQL computationally complete.

Some of the ODMG OQL fonctionnalities or specificities are not yet implemented:

1. the **group by/having** operator,
2. the **order by** operator is more restrictive than in the ODMG specifications,
3. contrary to ODMG OQL, it is necessary to put parenthesis to call a function or method with takes no arguments,
4. contrary to ODMG OQL, the **||** operator does means string concatenation. It is the logical or operator. This will be changed in a future version.

## 4 Language Concepts

The basic entity in OQL is the atom. An atom is the result of the evaluation of any expression. Atoms are manipulated through expressions.

Although OQL is not fully an expression language as some valid constructs, such as flow controls, are not expressions, the expression is the basic concept of OQL. The non-expression constructs, such as selection and iteration statements, control the flow (or evaluation) of expressions. An expression is built from typed operands composed recursively by operators.

OQL is a typed language: each atom has a type. This type can be an OQL builtin type or can be derived from the schema type declarations.

OQL binds the EYEDB object model by providing a mapping between OQL builtin types and the EYEDB object model types. As in the EYEDB object model, OQL supports both object entities (with a unique object identifier) and literal entities (with no identifier). The concept of object/literal is orthogonal to the concept of type, this means that any type may have object instances and literal instances. For instance, one can have literal or object integers, literal or object collections. For instance, the following constructs produces a literal integer:

```

1; // OQL interpreter generates a literal integer
first(select x.age from Person x); // database produces a literal integer
                                   // bound to an atom of type integer

```

while the followings produce respectively a **Person** object and a literal collection of **Person** objects:

```

first(select x from Person x);
select x from Person x;

```

An EYEDB object is always bound in OQL to an atom of type **oid** or of type **object**. A literal is bound to the corresponding direct type in OQL using a trivial mapping. For instance, a literal entity of the EYEDB type **integer** is bound to an OQL atom of the OQL type **integer**; while an object entity of the EYEDB type **Person** is bound to an OQL atom of type **oid** or **object**.

We introduce now the OQL builtin types and the way that they are generated. OQL includes 15 builtin types as follows:

- **integer**
- **string**
- **float**
- **char**
- **boolean**
- **identifier**
- **set**
- **bag**
- **array**
- **list**

- **struct**
- **oid**
- **object**
- **null**
- **nil**

Some of these atoms can be expressed as terminals of the OQL grammar - for instance **integer**, **float**, **string** - others are generated using syntactic constructions such as function calls or specific constructions - for instance **list**, **bag**, **struct**.

We will introduced first the syntax of the atoms which can be expressed as terminals, then the way to produce non terminal atoms.

## 5 Language Syntax

### 5.1 Terminal Atom Syntax

To express the syntax of terminal atoms, we use the standard regular expression notation.

#### Integer Atom

Integers are coded on 64 bits.

The syntax for the integer type is one of the followings:

```
[0-9]+          decimal base
0x[0-9a-fA-F]+ hexadecimal
0[0-7]+        octal
```

The domain for the integer type is as follows:

<i>Minimal Value</i>	<i>Maximal Value</i>
-9223372036854775808	9223372036854775807

A few examples:

```
13940      // integer expressed in the decimal base
0x273f1    // integer expressed in the hexadecimal base
0x273F1    // integer expressed in the hexadecimal base
0100       // integer expressed in the octal base
```

#### Float Atom

The syntax for floating point atoms is one of the following regular expressions:

```
[0-9]+\.[0-9]+?
[0-9]+?+\.[0-9]+
[0-9]+\.[0-9]+?(e|E) [+]? [0-9]+([fF] | [lL])?
[0-9]+?+\.[0-9]+(e|E) [+]? [0-9]+([fF] | [lL])?
```

The domain for the float type is as follows:

<i>Minimal Value</i>	<i>Maximal Value</i>
4.94065645841246544e-324	1.79769313486231570e+308

A few examples:

```
1.
1.23
.3
0.3039
1e+10
2.e+112
```



```
1.2e-100
.234e-200
.234e-200f
.234e-200F
```

### String Atom

The syntax for the string type is as follows:

```
"([~"]|\\\\")*"
```

The following escape sequences are interpreted:

<i>Escape Sequence</i>	<i>Name</i>	<i>ASCII Name</i>
<code>\a</code>	alert	BEL
<code>\b</code>	backspace	BS
<code>\f</code>	form feed	FF
<code>\n</code>	newline	NL (LF)
<code>\r</code>	carriage return	CR
<code>\t</code>	horizontal tab	HT
<code>\v</code>	vertical tab	VT
<code>\\</code>	backslash	<code>\</code>
<code>\"</code>	double quote	<code>"</code>
<code>\'</code>	single quote	<code>'</code>
<code>\ooo</code>	octal number	<code>\ooo</code>

A few examples:

```
"hello"
"hello \"world\""
"this is a multi-lines\ntext\n"
"this text contains escape sequences: \007\v\f\n",
```

### Char Atom

The syntax for the char type is one of the followings:

```
'ascii character'
'[0-7+]'
'(\x|X)[0-9a-fA-F+]'
'a'
'b'
'f'
'n'
'r'
't'
'v'
```

A few examples:

```
'a'
'b'
'\n'
'\a'
'\007'
'\x50'
'\x5F'
```

### Boolean Atom

The syntax for a boolean atom is one of the followings:

```
true
false
```

### Identifier Atom

The syntax for an identifier atom is as follows:

```
[a-zA-Z\_$.#] [a-zA-Z\_$.0-9#]*
```

This means that an identifier must start with a letter, a “\_”, a “\$” or a “#” which may be followed by letters, digits, “\_”, “\$” and “#” characters.

For instance, the following words are some valid identifiers:

```
a
alpha
beta1
alpha_beta
$a
oql$maxint
oql#2
$
#
_1
```

Note that identifiers beginning by `oql$` or `oql#` are reserved for special use by the interpreter.

### Oid Atom

The syntax for an `oid` is as following:

```
[0-9]+:[0-9]+:[0-9]+:oid
```

Note that `oid` atoms are not typed directly by the user, but are produced by the database via the OQL interpreter. The following words are some syntactically valid atom `oids`:

```
123.2.33373:oid
82727272.1.292828282:oid
```

### Object Atom

The syntax for an atom `object` is as following:

```
[0-9a-fA-F]+:obj
```

Note that object atoms are not typed directly by the user, but are produced by OQL interpreter. The following words are some syntactically valid atom `objects`:

```
38383:obj
ea954:obj
```

### Null Atom

The `null` atom denotes an uninitialized value. Its type depends on the context. It can denote a uninitialized integer, float, char, string or oid.

The syntax for a null atom is one of the followings:

```
null
NULL
```

### Nil Atom

The `nil` atom denotes the empty atom.

The syntax for a nil atom is as follows:

```
nil
```

## 5.2 Non Terminal Atom Production

The other atoms - `sets`, `bags`, `arrays`, `lists` and `structs` - are non terminal atoms. This means that they cannot be generated using a simple lexical construct.

### List, Set, Bag and Array Atoms

To construct a collection atom - **list**, **set**, **bag** or **array** -, one may use the function *collection()* where *collection* denotes the collection type, for instance:

```
set(1, 2, 3)
list(1, "hello", "world")
array(2, 3, list(3893, -2, 'a'), 22)
bag(2, 2, 3, 4, 5, 12)
```

This is the simple way to construct such atoms, but as any other atoms, a collection atom may be produced by the OQL interpreter as the evaluation of a complex expression, for instance:

```
select x from Person x
```

produces an atom **bag** of objects.

### Struct Atom

The most direct way to construct a **struct** atom is as follows:

```
struct({identifier: expr})
```

For instance:

```
struct(a: 1)
struct(format: text, s: "this is the text")
struct(name: "john", age: 32, spouse: first(select Person))
```

## 5.3 Keywords

Any programming language has its own set of reserved words (keywords) that cannot be used as identifiers. For instance, the keyword “**if**” cannot be used as a variable in a C program.

OQL also has its own set of keywords. But OQL is one part among others in the information system: for instance, there are an Object Model, an Object Definition Language (ODL) and Language bindings. The Object Model does not introduce any keyword, while ODL has its own set of keywords which are different from the OQL keywords. For instance, a class can include an attribute whose name is “**if**” as it is not a ODL keyword. If one wants to access this attribute in OQL, using for instance the path expression “**x.if**”, we will get a syntax error. This is not acceptable.

We introduce in OQL (and in ODL) a way to neutralize any keyword: the token “**@**” used as a prefix keyword neutralizes the keyword and makes it a valid identifier. For instance, “**x.@if**”, denotes the attribute “**if**” of the instance “**x**”.

More generally, “**@identifier**” denotes the identifier “*identifier*” whether “*identifier*” is a keyword or not.

OQL introduces the following keywords:

add	all	append	array	as
asc	bag	bodyof	break	by
char	classof	contents	define	delete
desc	distinct	do	element	else
empty	eval	except	exists	false
float	for	from	function	group
having	ident	if	import	in
int	intersect	is	isset	like
list	mod	new	nil	not
oid	order	pop	print	push
refof	return	scopeof	select	set
string	struct	structof	suppress	then
throw	to	true	typeof	union
unset	unval	valof	where	while

## 5.4 Comments

In OQL, comments are identical to the C++:

- all characters after the token `//` until the end of the current line are ignored by the interpreter,
- all characters between the tokens `/*` and `*/` are ignored.

For instance:

```
1 + 2;           // this is a comment
a := "hello"; /* this is another
               comment */
```

## 5.5 Statements

A valid OQL construct is composed of a sequence of statements. Main of the statements are expression statements.

A statement can be one of the following:

- an expression statement,
- a selection statement, **if/else**,
- an iteration statement, **while**, **do/while**, **for**,
- a function definition statement, **function**
- a jump statement, **break**, **return**
- a compound statement,
- an empty statement.

The OQL expression sub-grammar is very close from the C grammar. The OQL grammar for the flow controls statement - **if/else**, **while**, **do/while**, **for** - is identical to the C grammar. The common operators of OQL and C have the same associativity and precedence.

## 5.6 Expression Statements

An expression statement is an expression following by a semicolon. Expressions are built from typed operands composed recursively by operators.

The syntax of an expression statement is as follows:

*expr* ;

where *expr* denotes any expression.

There are three main kinds of expressions: atomic expressions, unary expressions and binary expressions. Atomic expressions are composed of one terminal atom and no operators, unary expressions are composed of one operand and one operator, binary expressions are composed of two operands and one operator.

We divide the OQL expression family into several semantical sub-families according to their operators as follows:

- atomic expressions,
- arithmetic expressions,
- assignment expressions,
- auto increment & decrement expressions,
- comparison expressions,
- logical expressions,
- conditional lists,
- expression sequences,
- array deferencing,
- identifier expressions,
- path expressions,
- function call,
- method invocation,
- eval/unval operators,
- set expressions,
- object creation,
- object deletion,
- collection expressions,
- exception expressions,
- function definition expressions,
- conversion expressions,
- type information expressions
- query expressions,
- miscellenaous expressions

In the following sub-sections, we introduce all the OQL expression types using the following template presentation:

1. we present first, in an unformal way, the syntax and the semantics of the operators,
2. general formal information is presented in a first table:
  - operator(s)
  - type
  - syntax
  - operand types
  - functions
3. in an optionnal second table, we present all the valid operand combination and their result type. A comment about the function performed is added if necessary. This table is skipped in case of the operand type combination is unique or trivial.
4. in a last table, we introduce a few examples. The examples manipulating database objects use the schema that can be found in the directory  
\$EYEDBROOT/examples/common:

```
// person.odl

enum CivilState {
    Lady = 0x10,
    Sir  = 0x20,
    Miss = 0x40
};

class Address {
    attribute string street;
    attribute string<32> town;
    attribute string country;
};

class Person {
    attribute string name;
    attribute int age;
    attribute Address addr;
    attribute Address other_addrs[];
    attribute CivilState cstate;
    attribute Person * spouse inverse Person::spouse;
    attribute set<Car *> cars inverse owner;
    attribute array<Person *> children;

    int change_address(in string street, in string town,
                      out string oldstreet, out string oldtown);

    static int getPersonCount();
    index on name;
};

class Car {
    attribute string brand;
    attribute int num;
    Person *owner inverse cars;
};

class Employee extends Person {
    attribute long salary;
};
```

Expression types are gathered so to minimize the number of tables in this document.

## 5.7 Atomic Literal Expressions

Atomic literal expressions are expressions composed of a single terminal (or token or lexical unit) without any operator. They are also called primary expressions. These expressions have already been introduced in Section 5.1.

<i>General Information</i>	
Operator	<i>no operator</i>
Type	unary
Syntax	<i>terminal atom</i>
Operand Types	integer, float, char, string, boolean, identifier, null, nil, oid,
Result Type	same type as the operand

<i>Expression Examples</i>	
<i>expression</i>	<i>result</i>
1	1
2.	12.
'a'	'a'
"hello"	"hello"
alpha	value of alpha
true	true
83283.1.29292:oid	an error is raised in case of the oid is invalid. Otherwise the result is the input oid: 83283.1.29292:oid

## 5.8 Arithmetic Expressions

Arithmetic expressions gather the expressions used for any arithmetic computation: addition, multiplication, substraction, division, modulo, shift left and right, bitwise or, bitwise exclusive, bitwise and, bitwise complement. This Section introduced these operators with a special focus on the additive operator which is a multi-purpose operator.

### Additive Expression

The additive operator (i.e. `+`) is used for arithmetic addition of integers, floating point numbers and characters, and is also used for string concatenation, list or array concatenation and set or bag union. Its functionality depends on the type of its operands: it is a polymorphic operator. Note that the choice of its functionality is done at evaluation time, not at compile time. That means that the functionality of an expression such as `x + y` is unknown until the evaluation time. Depending on the dynamic type of the operands `x` and `y`, it can be a simple arithmetic addition, a string or list or array concatenation, a set or bag union or it can raise an error.

When used as a arithmetic operator and when the two operands have not the same type, one of the operands can be automatically promote to the type of the second one. The *promotion* mechanism is the same as in the C or C++ languages: `integer` may be promoted to `float`, `char` may be promoted to `float` or `integer`.

<i>General Information</i>	
Operator	<code>+</code>
Type	binary
Syntax	<i>expr + expr</i>
Commutative	yes
Operand Types	integer, float, char, string, list, bag, set, array
Result Type	see following table

Function	multi functions according to operands: arithmetic addition, string concatenation, list or array concatenation, set or bag union.
----------	---

<i>Possible Operand Combinations</i>			
<i>first operand type</i>	<i>second operand type</i>	<i>result type</i>	
integer	integer	integer	
integer	float	float	
char	char	integer	
char	integer	integer	
char	float	float	
float	float	float	
string	string	string	<i>string concatenation</i>
list	list	list	<i>list concatenation</i>
array	array	array	<i>array concatenation</i>
set	set	set	<i>set union</i>
bag	bag	bag	<i>bag union</i>

<i>Expression Examples</i>	
<i>expression</i>	<i>result</i>
1 + 2	3
1 + 2.	3.
2 + 2.3	4.3
'a' + 'b'	195
'a' + 1.2	98.200
"hello" + "world"	"helloworld"
list(1, 2, 3) + list(2, 3, 4)	list(1, 2, 3, 2, 3, 4)
set(1, 2, 3) + set(2, 3, 4)	set(1, 2, 3, 4)
1 + "hello"	<i>raises an error</i>
set(1, 2, 3) + list(2, 3, 4)	<i>raises an error</i>

### Multiplicative, Division and Minus Expressions

Multiplicative, division and minus expression syntax, semantics, associativity and precedence are quite identical to the corresponding C and C++ expressions. When operands have different types, promotionnal mechanisms are the same as for the additive operator.

<i>General Information</i>	
Operators	- * /
Type	binary
Syntaxes	<i>expr - expr</i> <i>expr * expr</i> <i>expr / expr</i>
Commutative	- : no * : yes / : no
Operand Types	integer, float, char
Result Type	see following table

Functions	- : subtract * : multiply / : divide
-----------	--

<i>Possible Operand Combinations</i>		
<i>first operand type</i>	<i>second operand type</i>	<i>result type</i>
integer	integer	integer
integer	float	float
integer	char	integer
char	char	integer
char	integer	integer
char	float	float
float	float	float
float	integer	float
float	char	float

<i>Expression Examples</i>	
<i>expression</i>	<i>result</i>
1 - 2	-1
3 * 2.	6.
2 * 'a'	194
'a' * 'b'	9506
1 / 2	0
1 / 2.	.5000
1. / 2	.5000
1. / 2	.5000
"hello" * "world"	raises an error
1 - "hello"	raises an error

### Shift, Mod, And, Or, XOr Expressions

Shift, modulo, and, or and xor expression syntax, semantics, associativity and precedence are quite identical to the corresponding C and C++ expressions. Operand types must be **integer** or **char** and the only possible type promotion is from **char** to **integer**.

<i>General Information</i>	
Operators	<< >> % &   ^
Type	binary
Syntaxes	<i>expr</i> << <i>expr</i> <i>expr</i> >> <i>expr</i> <i>expr</i> % <i>expr</i> <i>expr</i> & <i>expr</i> <i>expr</i>   <i>expr</i> <i>expr</i> ^ <i>expr</i>
Commutative	<< : no >> : no % : no & : yes   : yes



	<code>^</code> : yes
Operand Types	<code>integer</code> , <code>char</code>
Result Type	<code>integer</code>
Functions	<code>&lt;&lt;</code> : left shift <code>&gt;&gt;</code> : right shift <code>%</code> : modulo <code>&amp;</code> : bitwise and <code> </code> : bitwise or <code>^</code> : bitwise exclusive or

<i>Expression Examples</i>	
<i>expression</i>	<i>result</i>
<code>1 &lt;&lt; 4</code>	<code>16</code>
<code>100 &gt;&gt; 2</code>	<code>25</code>
<code>100 % 13</code>	<code>9</code>
<code>0xf12 &amp; 0xf</code>	<code>2</code>
<code>0xf12   0xf</code>	<code>3871</code>
<code>0xf12 ^ 0xf</code>	<code>3869</code>
<code>'b' % '9'</code>	<code>8</code>
<code>2 &lt;&lt; 1.2</code>	<i>raises an error</i>
<code>2 % 3.4</code>	<i>raises an error</i>
<code>2.1 % 3</code>	<i>raises an error</i>

### Sign Expressions

Sign expressions are the expressions using the unary operators `+` or `-`. The expression syntax, semantics, associativity and precedence are quite identical to the corresponding C and C++ expressions. These unary operators accept only `integer`, `char` and `float` operands.

<i>General Information</i>	
Operators	<code>+</code> <code>-</code>
Type	unary
Syntaxes	<code>+ expr</code> <code>- expr</code>
Operand Types	<code>integer</code> , <code>char</code> , <code>float</code>
Result Type	see following table
Functions	sign operator

<i>Possible Operand Combinations</i>	
<i>operand type</i>	<i>result type</i>
<code>integer</code>	<code>integer</code>
<code>float</code>	<code>float</code>
<code>char</code>	<code>integer</code>

<i>Expression Examples</i>	
<i>expression</i>	<i>result</i>
<code>+12</code>	<code>12</code>
<code>-100</code>	<code>-100</code>
<code>-123.4</code>	<code>-123.4</code>
<code>+'a'</code>	<code>97</code>
<code>-'a'</code>	<code>-97</code>

<code>+"hello"</code>	<i>raises an error</i>
<code>-null</code>	<i>raises an error</i>

### Complement Expressions

The complement operator performs a bitwise complement on its operand. The expression syntax, semantics, associativity and precedence are quite identical to the corresponding C and C++ expressions. This operator accepts only **integer** and **char** operands.

<i>General Information</i>	
Operator	<code>~</code>
Type	unary
Syntax	<code>~ expr</code>
Operand Types	<b>integer</b> , <b>char</b>
Result Type	<b>integer</b>
Functions	bitwise complement

<i>Expression Examples</i>	
<i>expression</i>	<i>result</i>
<code>~112</code>	<code>-113</code>
<code>~0</code>	<code>-1</code>
<code>~'a'</code>	<code>-98</code>
<code>~2.3</code>	<i>raises an error</i>
<code>~"hello"</code>	<i>raises an error</i>

## 5.9 Assignment Expressions

The expression syntax, semantics, associativity and precedence are quite identical to the corresponding C and C++ expressions except that the simple assignment operator in OQL is `:=` instead of `=` in C or C++. The left operand must be a left value. A left value is an OQL entity which is assignable: for instance any identifier or a valid path expression.

When the assignment is simple and when the left value is an identifier, no type checking on the right operand is done. For instance, `x := 10` and `x := "hello"` are always valid expressions. In the case of the left value is a path expression, the OQL interpreter checks that the type of the second operand matches the expected type of the first one. For instance if `p` denotes a **Person** instance, `p->age := 32` is certainly valid while `p->age := "hello"` raises a type check error.

When the assignment is combined with another operation (for instance, the `-=` operator), the left operand must be initialized and the interpreter checks that the left and right operand can be combined through the other operator.

For instance, the following constructs are valid:

```
a := 10;
```

```
a -= 20;
```

```
a := "hello";
```

```
a += " world";
```

```
p := first(select Person);
```

```
p.name := "johnny";
```

```
first(select Person.age = 0).name := "baby";
```

while these ones produce errors:

```
a := "hello";
```

```
a -= 20; // raises the error: operation 'string + integer' is not valid
```

```
a := list(1, 2);
```

```
a *= 2; // raises the error: operation 'list * integer' is not valid
```

```
unset b;
```

```

b += 20; // raises the error: uninitialized identifier 'b'

p := first(select Person);
p.age := "baby"; // raises the error: integer expected, got string

```

<i>General Information</i>	
Operators	:= *= /= %= += -= <<= >>= &=  = ^=
Type	binary
Syntaxes	<i>lvalue</i> := <i>expr</i> <i>lvalue</i> *= <i>expr</i> <i>lvalue</i> /= <i>expr</i> <i>lvalue</i> %= <i>expr</i> <i>lvalue</i> += <i>expr</i> <i>lvalue</i> -= <i>expr</i> <i>lvalue</i> <<= <i>expr</i> <i>lvalue</i> >>= <i>expr</i> <i>lvalue</i> &= <i>expr</i> <i>lvalue</i>  = <i>expr</i> <i>lvalue</i> ^= <i>expr</i>
Commutative	no
Operand Types	<i>leftvalue</i> on the left side and <i>any type</i> on the right side
Result Type	the type of the right operand
Functions	perform an operation and assignment

<i>Expression Examples</i>	
<i>expression</i>	<i>result</i>
a := 24	24
a += 12	36
a /= 2	18
a ^= 100	118
first(select Person).age := 38	38
"hello" := 4	raises an error (i.e. "hello" is not a left-value)
8 := 5	raises an error (i.e. 8 is not a leftvalue)
unset a; a += 20	raises an error (i.e. uninitialized identifier)

### 5.10 Auto Increment & Decrement Expressions

The expression syntax, semantics, associativity and precedence are quite identical to the corresponding C and C++ expressions. The operand must be an initialized left value of type **integer**, **char** or **float**. In case of the operand is a **char** atom, the result type is an **integer**. Otherwise, the result type is the type of the operand.

<i>General Information</i>	
Operators	<b>++</b> <b>--</b>
Type	unary
Syntaxes	<i>expr --</i> <i>expr ++</i> <i>++expr</i> <i>--expr</i>
Operand Type	<i>leftvalue</i> of type <b>integer</b> , <b>char</b> or <b>float</b> .
Result Type	see following table
Functions	<i>expr --</i> : post-decrementation <i>expr ++</i> : post-incrementation <i>++expr</i> : pre-incrementation <i>--expr</i> : pre-incrementation

<i>Possible Operand Combinations</i>	
<i>operand type</i>	<i>result type</i>
<b>integer</b>	<b>integer</b>
<b>float</b>	<b>float</b>
<b>char</b>	<b>integer</b>

<i>Expression Examples</i>		
<i>expression</i>	<i>result</i>	
<b>a := 1; a++</b>	<b>1</b>	<i>initially a equals 1; the result of the evaluation is 1 but after the evaluation, a equals 2</i>
<b>--a</b>	<b>0</b>	
<b>a++</b>	<b>0</b>	<i>a equals 1 after the evaluation</i>

## 5.11 Comparison Expressions

The expression syntax, semantics, associativity and precedence are quite identical to the corresponding C and C++ expressions except that the equal operator could be either **==** or **=**.

### Equal and NotEqual Expressions

Operands may have any type at all. If the type of the operands differ (modulo the type promotion mechanisms for numbers), the result of the expression *operand1 == operand2* is always **false** while the result of *operand1 != operand2* is always **true**.

<i>General Information</i>	
Operators	<b>==</b> <b>!=</b>
Type	binary
Syntaxes	<i>expr == expr</i> <i>expr != expr</i>
Commutative	yes
Operand Types	<i>any type</i>
Result Type	<b>boolean</b>
Functions	equal not equal

When operands are number of different types, an automatic promotion is done to the more precise type.

<i>Expression Examples</i>	
<i>expression</i>	<i>result</i>
<code>1 == 1</code>	<code>true</code>
<code>1 == 1.0</code>	<code>true</code>
<code>1 != 2</code>	<code>true</code>
<code>1 == 2</code>	<code>false</code>
<code>1 == "hello"</code>	<code>false</code>
<code>"hello" == "hello"</code>	<code>true</code>
<code>list(1, 2, 3) == list(1, 2, 3)</code>	<code>true</code>
<code>set(1, 3, 2) == set(1, 2, 3)</code>	<code>true</code>

### Less and Greater Expressions

The comparison operators `<`, `<=`, `>` and `>=` are multi-purpose operators: they are used for integer, floating point number and character comparison, but also for list or array term-to-term comparison and for set or bag inclusion. Their functionality depends on the type of its operands: they are polymorphic operators. Note that the choice of the functionality is done at evaluation time, not at compile time. That means that the functionality of an expression such as `x < y` is unknown until the evaluation time. Depending on the dynamic type of the operands `x` and `y`, it can be an arithmetic comparison (if `x` and `y` are numbers), a alpha-numeric comparison (if `x` and `y` are strings), a term-to-term ordered collection comparison (if `x` and `y` are lists or arrays) or a set or bag inclusion comparison (if `x` and `y` are sets or bags).

While arithmetic and alpha-numeric comparisons are trivial and do not need any supplementary explanations, the term-to-term ordered collection comparisons needs to be detailed.

The general algorithm for this fonctionnality is as follows:

1. let `l1` and `l2` two OQL ordered collections, containing respectively `l1_cnt` and `l2_cnt` atoms.
2. let `op` one of the following polymorphic comparison operators: `<` `<=` `>` `>=`,
3. `l1 op l2` is `true` if and only if all the following conditions are realized:
  - (a) `l1` and `l2` must be of the same collection type,
  - (b) `l1_cnt op l2_cnt` or `l1_cnt` equals `l2_cnt`
  - (c) for each atom `l1[i]` with `i` in `[i, l1_cnt]`, `l1[i] op l2[i]`

For instance:

```
list(1, 2) <= list(0, 2) is true
list(1, 2) <= list(3) is false
list(1, 2) <= list(3) is false
list("aaa", 4) < list("bbbb", 8) is true
list("aaa", 4, list(1, 2)) < list("b", 8, list(2, 3)) is true
list(set(2, 4), 3) < list(set(4, 2, 3), 4) is true
list(2, 3) < list("hello", 2) raises an error
list(2, 3) < array(2, 4) raises an error
```

Note that the fact that `l1 <= l2` is `false` does not implice that `l1 > l2` is `true`. Indeed, `list(2, 3) < list(1, 3, 2)` and `list(1, 3, 2) >= list(2, 3)` are `false`.

<i>General Information</i>	
Operators	<code>&lt;</code> <code>&lt;=</code> <code>&gt;</code> <code>&gt;=</code>
Type	binary
Syntaxes	<code>expr &lt; expr</code> <code>expr &lt;= expr</code> <code>expr &gt; expr</code> <code>expr &gt;= expr</code>

Commutative	no
Operand Types	integer, float, char, string, list, array, set, bag
Result Type	boolean
Functions	the function depends on the operands: < : less than <i>or</i> is included in <= : less than or equal <i>or</i> is included in or equal > : greater than <i>or</i> contains >= : greater than or equal <i>or</i> contains or equal

<i>Possible Operand Combinations</i>			
<i>first operand type</i>	<i>second operand type</i>	<i>result type</i>	<i>comments</i>
integer, char, float	integer, char, float	boolean	performs an arithmetic comparison
string	string	boolean	performs an alpha-numeric comparison
set	set	boolean	performs an inclusion comparison
bag	bag	boolean	performs an inclusion comparison
set	bag	boolean	performs an inclusion comparison: the <b>set</b> operand is converted to a <b>bag</b>
bag	set	boolean	performs an inclusion comparison: the <b>set</b> operand is converted to a <b>bag</b>
list	list	boolean	performs a term-to-term polymorphic (i.e. numeric, alpha-numeric or inclusion) comparison
array	array	boolean	performs a term-to-term polymorphic comparison

Note that in case of different operand types, an automatic promotion is done to the more precise type.

<i>Expression Examples</i>	
<i>expression</i>	<i>result</i>
1 < 2	true
1 >= 2	false
2. <= 2	true
"hello" < "world"	true
"hello" >= "world"	false
list(1, 2) < list(2, 3)	true
list(1, 2) < list(0, 3)	false
list(1, 2) < list(0, 3, 2)	false
list(0, 3, 2) >= list(1, 2)	false
list(1, 2) < list(2, 3, 3)	true
list(1, 2) < list(0)	false
set(1, 2) < set(2, 4, 44)	false
set(1, 2) < set(2, 1, 44)	true
"hello" >= 3	raises an error
list(1, 2) < array(2, 4, 44)	raises an error
set(1, 2) < bag(2, 1, 44)	raises an error

## Regular Expression Operators

OQL provides the ODMG OQL regular expression operator **like** plus four other ones. These four extra operators are based on the regular expression UNIX library. So, the syntax of the regular expression are the same as that used by the well known UNIX tools **grep**, **sed**, and so on. All the regular expression operators takes two string operands: the first one is the string to compare, the second one is the regular expression. So, these operators are not commutative. These operators provide the following functionalities:

- like** : ODMG OQL operator. Returns **true** if the first operand matches the regular expression. Otherwise **false** is returned. The regular expression is an SQL regular expression where, for instance, % and \_ are wildcard characters.
- ~** : This operator has the same functionality as the **like** operator but the regular expression has the UNIX syntax.
- ~** : Returns **true** if the first operand matches the regular expression in a case insensitive way. Otherwise **false** is returned.
- !~** : Returns **true** if the first operand does not match the regular expression. Otherwise **false** is returned.
- !~~** : Returns **true** if the first operand does not match the regular expression in a case insensitive way. Otherwise **false** is returned.

*Note that the operator **like** uses currently the UNIX form of regular expressions instead of the SQL form. It will become ODMG/SQL compliant in a next version.*

General Information		
Operators	~ ~~ !~ !~~ like	
Type	binary	
Syntaxes	<i>expr ~ expr</i> <i>expr ~~ expr</i> <i>expr !~ expr</i> <i>expr !~~ expr</i> <i>expr like expr</i>	
Commutative	no	
Operand Types	<b>string</b>	
Result Type	<b>boolean</b>	
Functions	~ : matches the regular expression ~ : matches the regular expression, case insensitive !~ : does not match the regular expres- sion !~~ : does not match the regular expres- sion, case insensitive like : matches the regular expression	

Expression Examples	
<i>expression</i>	<i>result</i>
"hello" ~ "LL"	false
"hello" ~~ "LL"	true
"hello" ~ "^LL"	false
"hello" ~ "^h"	true
"hello" !~ "^h"	false
"hello" ~ ".*ll.*"	true
".*ll.*" ~ "hello"	false because regular ex- pression should be on the right

## 5.12 Logical Expressions

OQL provide three logical expressions which can take two form each. The logical or operator is `||` or `or`. The logical and operator is `&&` or `and`. The logical not operator is `!` or `not`.

The expression syntax, semantics, associativity and precedence are quite identical to the corresponding C and C++ expressions. Note that the ODMG operator “`||`” denotes the string concatenation.

As for C and C++, the OQL interpreter performs a lazy evaluation:

- *expr1 || expr2*  
*expr2* is not evaluated if *expr1* is evaluated to **true**.
- *expr1 && expr2*  
*expr2* is not evaluated if *expr1* is evaluated to **false**.

<i>General Information</i>	
Operators	<code>  </code> <code>&amp;&amp;</code>
Type	unary, binary
Syntaxes	<i>expr    expr</i> <i>expr or expr</i> <i>expr &amp;&amp; expr</i> <i>expr and expr</i> <i>! expr</i> <i>not expr</i>
Operand Type	<b>boolean</b>
Result Type	<b>boolean</b>
Functions	logical or logical and logical not

<i>Expression Examples</i>	
<i>expression</i>	<i>result</i>
<code>true    false</code>	<b>true</b>
<code>false    false</code>	<b>false</b>
<code>true &amp;&amp; false</code>	<b>false</b>
<code>1 == 2    3 == 4</code>	<b>false</b>
<code>1 == 2    3 == 3</code>	<b>true</b>
<code>1 == 2 or 3 == 3</code>	<b>true</b>
<code>1 == 2    "hello" == "hello"</code>	<b>true</b>
<code>(1 == 2    2 == 2) &amp;&amp; (a = "hello")</code>	returns <b>true</b> if <code>a</code> equals "hello". <b>false</b> otherwise
<code>1    3 == 3</code>	raises an error: boolean expected got integer
<code>!3</code>	raises an error: boolean expected got integer
<code>!(1 == 1)</code>	<b>false</b>
<code>not(1 == 1)</code>	<b>false</b>

## 5.13 Conditional Expression

The unique conditional expression operator is `?:`. The expression syntax, semantics, associativity and precedence are quite identical to the corresponding C and C++ expressions. The first operand must be an boolean and the two others may be of any type. Contrary to C and C++, the two last operands does not need to be of the same type.

<i>General Information</i>
----------------------------



Operator	<code>?:</code>
Type	ternary
Syntaxe	<i>expr ? expr : expr</i>
Operand Types	first operand is boolean, others are any type
Result Type	type of the evaluated operand
Functions	conditional evaluation: evaluates and returns the second operand if the first operand is true; otherwise evaluates and returns the second operand

<i>Expression Examples</i>	
<i>expression</i>	<i>result</i>
<code>true ? "hello" : "world"</code>	<code>"hello"</code>
<code>true ? 2.3 : "world"</code>	<code>2.3</code>
<code>1+1 == 2 ? (a := 3.1415926535) : nil</code>	<code>3.1515926535</code>
<code>1 ? 3 : nil</code>	<i>raises an error: boolean expected, got integer.</i>

### 5.14 Expression Sequences

The expression sequence operator - also called comma sequencing - expression syntax, semantics, associativity and precedence are quite identical to the corresponding C and C++ expressions. This operator , takes two operands: it evaluates both of them and returns the second one.

<i>General Information</i>	
Operator	<code>,</code>
Type	binary
Syntaxe	<i>expr , expr</i>
Commutative	no
Operand Types	<i>any type</i>
Result Type	type of the second operand
Functions	evaluates the first operand, then the second one. Returns the evaluation of the second one.

<i>Expression Examples</i>	
<i>expression</i>	<i>result</i>
<code>true, "hello"</code>	<code>"hello"</code>
<code>a := 2, 4</code>	<code>4</code> ( <i>note that a equals 2</i> )
<code>b := 10, a := b+1</code>	<code>11</code>

### 5.15 Array Deferencing

OQL provides polymorphic single and range deferencing. The single deferencing is used to get one element in an ordered collection or a character in a string or one element in a non-collection array. The range deferencing is used to get several elements.

The deferencing of ordered collections is introduced in more details in Section 5.24.

## Single Deferencing

The single deferencing operator takes as its first operand an atom of type **string**, an indexed (or ordered) collection (**list** or **array**) or a non-collection array. The second operand must be of type **integer**. Depending on the type of the first operand, the returned atom is as follows:

1. if the first operand is a string, the returned atom is the *#expr* character of the string where *expr* denotes the second operand. If *expr* is equal to the length to the string the character `'\000'` is returned. If it greater than the length the string an *out of bounds* error is raised.
2. for an ordered collection, the returned atom is the *#expr* item of the collection. If *expr* is greater than or is equal to the size of the collection, an *out of bounds* error is raised.
3. if the first operand is an non-collection array, the returned atom is the *#expr* item of the array. If *expr* is greater than or is equal to the size of the array, an *out of bounds* error is raised.

The single deferencing operator may be used as a left value, that means that a single deferencing expression is assignable. For instance the sequence of statements:

```
s := "hello";
s[1] := 'E';
s[4] := '0';
```

set the variable **s** to **"hE1l0"**.

The single deferencing operator may be used everywhere in a path expression. For instance, `first(select Person).other_addrs[2].street` denotes the character **#3** of the **street** attribute in the **#2 other\_addrs** non-collection array attribute of the first **Person** instance.

General Information	
Operator	<code>[]</code>
Syntaxe	<i>expr</i> [ <i>expr</i> ]
Type	binary
Commutative	no
Operand Types	first operand: <b>string</b> , indexed collection ( <b>list</b> or <b>array</b> ) or non-collection array, second operand: <b>integer</b>
Result Type	<b>char</b> if first operand is a string, otherwise type of the returned item in the indexed collection or non-collection array.
Functions	[ <i>expr</i> ] : returns the character (or item in the indexed collection or in the non-collection array) number <i>expr</i>
Note	this operator may be used in the composition of a left value.

Expression Examples		
<i>expression</i>	<i>result</i>	
"hello"[0]	'h'	
a := "hello"; a[1]	'e'	
a[3]	l	
a[6]	raises an error	
a[0] := 'H'	'H'	a equals "Hello"
list(1, 2, "hello", 4)[3]	"hello"	
list(1, 2, "hello", 4)[4]	raises an error	
first(select Person).name[2] := 'X'	'X'	

## Range Deferencing

The range deferencing operators, `[:]` and `[?]`, takes as their first operand an atom of type **string**, an indexed (or ordered) collection (**list** or **array**) or a non-collection array. The other operands must be of type **integer**. The `[?]` may have also an unordered collection (**set** or **bag** as its first operand.

The operator syntax and semantics are as follows:

- `expr [expr1:expr2]`
  1. if the first operand is a string, the returned atom is a list composed of the characters between the `#expr1` and the `#expr2` characters of the string. If `expr1` is less than zero or if `expr2` is greater than the length of the string an *out of bounds* error is raised.
  2. for an ordered collection, the returned atom is a list composed of the items between the `#expr1` and the `#expr2` items of the collection. If `expr1` is less than zero or if `expr2` is greater than or is equal to the size of the collection, an *out of bounds* error is raised.
  3. if the first operand is an non-collection array, the returned atom is a list composed of the items between the `#expr1` and the `#expr2` items of the array. If `expr1` is less than zero or if `expr2` is greater than or is equal to the size of the collection, an *out of bounds* error is raised.
- `expr [?]`
  1. if the first operand is a string, the returned atom is a list composed of all the characters of the string, including the last character `'\000'`.
  2. for an ordered or an unordered collection, the returned atom is a list composed of all the items of the collection. If the collection is a list, the list itself is returned. If the collection is an array, this operator has the same functionality as the `listtoarray` library function.
  3. if the first operand is an non-collection array, the returned atom is a list composed of the all the items of the array.

Contrary to the single deferencing operator, the range deferencing operator cannot be used as a left value.

The range deferencing operators may be used everywhere in a path expression. For instance, `first(select Person).children[?].name[?]`, denotes the list of all characters of the **name** attribute in all the **children** of the first **Person** instance.

General Information	
Operators	<code>[:]</code> <code>[?]</code>
Syntaxes	<code>expr [expr : expr ]</code> <code>expr [?]</code>
Type	ternary or unary
Operand Types	first operand: <b>string</b> or indexed collections ( <b>list</b> or <b>array</b> ), second operand and third operand: <b>integer</b>
Result Type	a list of <b>char</b> if first operand is a string, otherwise a list of returned items in the indexed collection or non-collection array.
Functions	<code>[expr1:expr2]</code> : returns a lits of characters (or items in collection) indexed from <code>expr1</code> to <code>expr2</code> <code>[?]</code> : returns a list of all characters (or items in collection)

Expression Examples	
expression	result
<code>"hello"[0:2]</code>	<code>list('h', 'e', 'l')</code>
<code>"hello"[?]</code>	<code>list('h', 'e', 'l', 'l', 'o', '\000')</code>
<code>list(1, 2, "hello", 4)[2:3]</code>	<code>list("hello", 4)</code>
<code>array(1, 2, "hello", 4)[?]</code>	<code>list(1, 2, "hello", 4)</code>

<code>first(select Person).name[?]</code>	<code>list('j', 'o', 'h', 'n', '\000')</code>
<code>list(select class.type = "user") [0:4].name</code>	<code>list("Employee", "Address", "Person")</code>

## 5.16 Identifier Expressions

We call an identifier expression an unary or binary expression whose operands must be identifiers. There are height identifier operators: `::`, `isset`, `unset`, `&` (identical to `refof`), `*` (identical to `valof`), `scopeof`, `push` and `pop`.

As all these operators take identifiers as their operands, we skip the second table (operand combinations) while introducing these operators.

### `::` Operator

The `::` unary/binary operator (called scope operator) is used to define a global or particular scope for a variable.

The unary version of this operator denotes a global scope. For instance, `::alpha` denotes the global variable `alpha`. In the body of a function, identifiers denote local variables; outside the body of a function identifiers denote global variables, that means that, in this context, the global scope operator is not mandatory. If one wants to use a global variable in the body of a function, the global scope operator is mandatory. Refer to the Function Definition Statement Section for more information about local function variables.

The binary version of this operator denotes a particular scope. For instance, `Person::checkName` denotes the class attribute or method of the class `Person`.

*note:* class (or static) attributes are not currently well supported by the OQL interpreter. Class attributes are only supported in some specific query expressions (refer to the Query Expression Section).

<i>General Information</i>	
Operator	<code>::</code>
Syntax	<code>:: identifier</code> <code>identifier::identifier</code>
Type	unary and binary
Operand Types	<code>identifier</code>
Result Type	value of the identifier if used as a right value; identifier reference if used as a left value
Function	defines a global or particular scope for the identifier.

<i>Expression Examples</i>	
<i>expression</i>	<i>result</i>
<code>::a</code>	the value of the global variable <code>a</code>
<code>::alpha := 1</code>	sets the value of the global variable <code>alpha</code> to 1, returns 1
<code>Person::checkName("wayne")</code>	calls the class method <code>checkName</code> in the class <code>Person</code>
<code>2::alpha</code>	raises an error

### `isset` Operator

The `isset` operator is used to check whether a variable is already set or not. It returns `true` if the variable is set, `false` otherwise.

<i>General Information</i>
----------------------------

Operator	<b>isset</b>
Syntax	<b>isset</b> <i>identifier</i>
Type	unary
Operand Type	<b>identifier</b>
Result Type	<b>boolean</b>
Function	evaluated to <b>true</b> if the identifier is set, <b>false</b> otherwise

<i>Expression Examples</i>	
<i>expression</i>	<i>result</i>
<b>isset</b> oql\$variables	<b>true</b>
<b>isset</b> a	returns <b>true</b> if <b>a</b> is set, <b>false</b> otherwise
<b>isset</b> 1	<i>raises an error</i>

### unset Operator

The **unset** operator is used to unset a variable. It returns the **nil** atom.

<i>General Information</i>	
Operator	<b>unset</b>
Syntax	<b>unset</b> <i>identifier</i>
Type	unary
Operand Type	<b>identifier</b>
Result Type	<b>nil</b>
Function	unset the identifier

<i>Expression Examples</i>	
<i>expression</i>	<i>result</i>
<b>unset</b> a	<b>nil</b>
<b>unset</b> ::a	<b>nil</b>
<b>unset</b> 2	<i>raises an error</i>

### refof Operator

The **&** (identical to **refof**) operator is used to get the reference of an identifier. This operator is essentially used when one calls a function or method which updates one or more given parameters. For instance, let the function **swap(x, y)** which swaps the value of its two parameters. One needs to give the references of the variables that one wants to swap. For instance:

```
i := "ii";
j := "jj";

swap(&i, &j);
```

After the call to **swap**, the variable **i** equals **jj** while the variable **j** equals **ii**. The reverse operator **\*** (described following section) is used in the swap function.

<i>General Information</i>	
Operator	<b>refof</b> <b>&amp;</b>
Syntax	<b>refof</b> <i>identifier</i> <b>&amp;</b> <i>identifier</i>
Type	unary

Operand Type	<b>identifier</b>
Result Type	<b>identifier</b>
Function	evaluates the expression to the identifier reference; returned an identifier atom

<i>Expression Examples</i>	
<i>expression</i>	<i>result</i>
<b>&amp;alpha</b>	<b>alpha</b>
<b>refof alpha</b>	<b>alpha</b>

### valof Operator

The **\*** (identical to **valof**) operator is used to get the value of the identifier pointed by a reference. For instance, after the two following expressions:

```
alpha := 1;
ralpha := &alpha;
```

**\*ralpha** equals 1.

This operator may be used in the composition of a left value, for instance:

```
alpha := 1;
ralpha := &alpha;
*ralpha := 2; // now, alpha equals 2
*ralpha += 8; // now, alpha equals 10
```

But this operator is essentially used in the body of functions or methods which update one or more given parameters, for instance, the function **swap** described in the previous section is as follows:

```
function swap(x, y) {
    v := *x;
    *x := *y;
    *y := v;
}
```

<i>General Information</i>	
Operator	<b>valof</b> <b>*</b>
Syntax	<b>valof</b> <i>identifier</i> <b>*</b> <i>identifier</i>
Type	unary
Operand Type	<b>identifier</b>
Result Type	value of the identifier
Function	returns the value of the identifier denotes by the operand

<i>Expression Examples</i>	
<i>expression</i>	<i>result</i>
<b>*alpha</b>	if <b>alpha</b> value is an atom identifier <i>x</i> , returns the value of <i>x</i> , otherwise an error is thrown
<b>x := 12; alpha := &amp;x; *x</b>	<b>*x</b> returns 12

**scopeof Operator**

The `scopeof` operator returns the string "global" or "local" depending whether the identifier is global or local.

<i>General Information</i>	
Operator	<b>scopeof</b>
Syntax	<b>scopeof</b> <i>identifier</i>
Type	unary
Operand Type	<b>identifier</b>
Result Type	<b>string</b>
Function	returns the scope of the identifier.

<i>Expression Examples</i>	
<i>expression</i>	<i>result</i>
<b>scopeof ::alpha</b>	returns "global" for any <b>alpha</b> if it set; otherwise an error is thrown
<b>scopeof alpha</b>	returns "global" or "local" depending on the context.

**push Operator**

The `push` operator is used to push an identifier on a new local table. This operator is rarely used.

<i>General Information</i>	
Operator	<b>push</b>
Syntax	<b>push</b> <i>identifier</i> <b>push</b> <i>identifier</i> := <i>expr</i>
Type	unary and binary
Operand Types	first operand <b>identifier</b> , optionnal second operand: <i>any type</i>
Result Type	identifier or any type in case of an assignment
Function	push the identifier on to the symbol table stack. An assignment can be performed at the same time. Returns the identifier or the value of the expression assignment.

<i>Expression Examples</i>	
<i>expression</i>	<i>result</i>
<b>push a</b>	pushes <b>a</b> on a new local symbol table.
<b>push a := 10</b>	pushes <b>a</b> on a new local symbol table and assigns its value to 10

**pop Operator**

The `pop` operator is used to pop an identifier from a local table. It is used after a push. For instance:

```
a := "hello";
a;           // a equals "hello"

push a := 10;
a;           // a equals 10
```

```
pop a;
a;           // a equals "hello"
```

<i>General Information</i>	
Operator	<b>pop</b>
Syntax	<b>pop</b> <i>identifier</i>
Type	unary
Operand Type	<b>identifier</b>
Result Type	the type of the value of the identifier
Function	pop the identifier from the symbol table stack

<i>Expression Examples</i>	
<i>expression</i>	<i>result</i>
<b>pop a</b>	returns the value of <b>a</b> if it is set; otherwise an error is returned

## 5.17 Path Expressions

The path expression operator `->` (identical to `.`) is used to *navigate* from an object and read the right data one needs. This operator enables us to go inside complex objects, as well as to follow simple relationships. For instance, if **p** denotes a **Person** instance, **p.spouse** denotes the **spouse** attribute of this person.

The more complex expression **p.spouse.address.street** denotes the **street** in the **address** of **spouse** of the person **p**. This notation is very intuitive because it looks like the well known **C**, **C++** and **Java** syntaxes.

The path expression operator may composed a left value, for instance:

```
p.spouse.name := "mary";
```

set the **name** of the **spouse** of the person **p** to **mary**.

This operator may be combined with the array deferencing operators, for instance:

```
p.spouse.name[2];
p.spouse.name[2] := 'A';
p.spouse.other_addrs[2].street[3] := 'C';
p.spouse.children[?];
p.spouse.children[?].name;
```

The path expression operator may be also used to navigate through **struct** atom, for instance: **(struct(a : 1, b : "hello")).b** returns **"hello"**. Note that because of the precedence of operators, parenthesis are necessary around the literal **struct** construct.

Finally, the path operator may be applied to a collection; in this case a collection of the same type of this operand is returned. For instance:

**(select Person).name** returns a **bag** of **string**.

**(select distinct Person).age** returns a **set** of **int**.

Note that the path expression operator is used frequently in the query expressions as shown in a next section.

<i>General Information</i>	
Operators	<code>.</code> <code>-&gt;</code>
Syntaxes	<i>expr . expr</i> <i>expr -&gt; expr</i>
Type	binary
Operand Types	first operand: <b>oid</b> or <b>object</b> , second operand: <b>identifier</b>



Result Type	type of the attribute denoted by the second operand
Functions	returns the attribute value denoted by second operand of the object denoted by the first operand The first operand must denote an EYEDB instance (object or literal) of an agregat including the attribute denoted by the second operand.
Note	these two operators are identical

<i>Expression Examples</i>		
<i>expression</i>	<i>result</i>	<i>comments</i>
<b>p-&gt;name</b>	<i>the value of attribute name in the object denoted by p</i>	<b>p</b> must denote an EYEDB instance (object or literal) of an agregat including the attribute <b>name</b>
<b>first(select x Person x from x.lastname = "wayne")-&gt;lastname</b>	"wayne"	

## 5.18 Function Call

OQL allows one to call an OQL function with or without parameters. The operator for function call is ().

A function call may be the first term of a path expression, for instance: **first(select Person)->name**.

Contrary to the method invocation, there are no function overloading mechanisms: that means, that one cannot have differents functions with the same name and a different signature. To take benefit of the overloading mechanisms, one must use methods.

*Note:* contrary to the ODMG 3 specifications, one currently needs to use parenthesis to invoke a method even if the method has no arguments.

<i>General Information</i>	
Operator	()
Syntaxe	<i>expr (expr_list)</i>
Type	n-ary
Operand Types	first operand: <b>identifier</b> , other operands: <i>any type</i>
Returned type	type of the returned atom by the function call
Functions	calls the OQL function denoted by the first operand using the other operands as arguments. The number of operands must be equal to the number of arguments of the OQL function plus one

<i>Expression Examples</i>	
<i>expression</i>	<i>result</i>
<b>fact(10)</b>	3628800
<b>fact(fact(3))</b>	720
<b>toUpper("hello world")</b>	"HELLO WORLD"
<b>toUpper("hello") + "world"</b>	"HELLOworld"
<b>interval(1, 5)</b>	list(1, 2, 3, 4, 5)
<b>swap(&amp;i, &amp;j)</b>	nil
<b>first(select Person).spouse.name</b>	"mary"

## 5.19 Method Invocation

### Instance Method Invocation

OQL allows one to call a instance method with or without parameters. The method can be written in C++ or in OQL. As in C++, method calls use a combination of the path expression operator and the function call operator.

As in C++ or Java, methods can be overloaded: that means that one can have different methods with the same name and a different signature or different methods with the same name and the same signature in a class hierarchy. The choice of the method to invoke is done at evaluation time not at compile time. For instance let two methods `Person` `Person::f(in int, in int)` and `int` `Person::f(in float, in string)`, the method to be invoked in the expression `p->f(x, y)` is decided at evaluation time according to the true types of `x` and `y`:

```
p := first(select Person);

x := 1; y := 2;

p->f(x, y); // X::f(in int, in int) is invoked
p->f(x, y)->name; // this is valid because p->f(x, y) returns a Person

x := 1.3; y := "hello";

p->f(x, y); // X::f(in float, in string) is invoked
p->f(x, y)->name; // this is not valid because p->f(x, y) returns an integer
```

A major contribution of object orientation is the possibility of manipulating polymorphic objects and thanks to the late binding mechanism to carry out generic actions on the elements of these objects.

For instance, let the two methods `void` `Person::doit(in int)` and `void` `Employee::doit(in int)`, the method to be invoked in the expression `p->doit(x)` is decided at evaluation time according to the true type of `p`:

```
p := new Person();
p->doit(1); // Person::doit(in int) is invoked

p := new Employee();
p->doit(1); // Employee::doit(in int) is invoked
```

To invoke a method, the following conditions must be realized:

1. the object or oid on which the method is applied must be an instance of a class, for instance `X`.
2. the name of the invoked method, the number and the type of arguments must be compatible with an existing method in the class `X`,
3. the result type must match the expected type in the expression.

For instance, let the methods `int` `compute(in int, in float)` and `int` `compute(in int, in float, in int[], out string)` in the class `X`. To invoke the first method on an instance of `X`, one needs to apply the method `compute` to an instance of `X` with one integer and one float, for instance:

```
x := new X();

x.compute(1, 2.3);
x.compute(a := fact(10), float(fib(10)));
```

To invoke the second method on an instance of `X`, one needs to apply the method `compute` to an instance of `X` with an integer, a float, an ordered collection of integer and a reference to a variable, for instance:

```
x.compute(1, 23.4, list(1, 2, 3, 4), &a);
```

The following table shows the mapping (which defines the compatibility) between the ODL and the OQL types.

ODL/OQL Mapping	
ODL Type	OQL Type
<code>in int16</code>	<code>integer</code>
<code>out int16</code>	<code>identifier</code>
<code>inout int16</code>	<code>identifier initialized to an integer</code>

in int32	integer
out int32	identifier
inout int32	identifier initialized to an integer
in int64	integer
out int64	identifier
inout int64	identifier initialized to an integer
in byte	char
out byte	identifier
inout byte	identifier initialized to a char
in char	char
out char	identifier
inout char	identifier initialized to a char
in string	string
out string	identifier
inout string	identifier initialized to a string
in float	float
out float	identifier
inout float	identifier initialized to a float
in oid	oid
out oid	identifier
inout oid	identifier initialized to a oid
in object *	oid of any class
out object *	identifier
inout object *	identifier initialized to an oid of any class
in X * (X denotes a class instance)	oid of class X
out X * (X denotes a class instance)	identifier
inout X * (X denotes a class instance)	identifier initialized to a oid of class X
in X *[] (X denotes a class instance)	ordered collection of oid of class X
out X *[] (X denotes a class instance)	identifier
inout X *[] (X denotes a class instance)	identifier initialized to an ordered collection of oid of class X
in X[] (X denotes any ODL type)	ordered collection of atoms bound to X
out X[] (X denotes any ODL type)	identifier
inout X[] (X denotes any ODL type)	identifier initialized to an ordered collection of atoms bound to X

*Note:* contrary to the ODMG 3 specifications, one currently needs to use parenthesis to invoke a method even if the method has no arguments.

General Information	
Operators	. ->
Syntaxes	<i>expr</i> . <i>expr</i> ( <i>expr-list</i> ) <i>expr</i> -> <i>expr</i> ( <i>expr-list</i> )
Type	n-ary
Operand Types	first operand: <b>oid</b> or <b>object</b> , second operand: identifier, other operands: any type
Result Type	type of the atom returned by the method call
Functions	invokes the method denoted by the second operand applied to the object denoted by the first operand, using the other operands as arguments.

	The first operand must denote an EYEDB instance (object or literal) of an agregat including the method whose name is the second operand. The number of arguments and the type of arguments must match one of the methods included in the class of the object denoted by the first operand.
Note	these two operators are identical

<i>Expression Examples</i>		
<i>expression</i>	<i>result</i>	<i>comments</i>
<code>p-&gt;getId()</code>	<i>the value of the oid of object denoted by p</i>	as <code>getId()</code> is a native method of the class <code>object</code> , each object can call this method
<code>img-&gt;compute(1, 2.3)</code>	<i>the value returned by the method call</i>	the first operand must denote an EYEDB instance (object or literal) of an agregat including the method whose name is <code>compute</code>
<code>first(select Person.name = "wayne").getSpouse()</code>	<i>the value returned by the method call</i>	

### Class Method Invocation

OQL allows one to call a class method with or without parameters. The method can be written in C++ or in OQL. As in C++, method calls use a combination of the scope operator and the function call operator. To invoke a class method, the following conditions must be realize:

1. the name of the invoked method, the number and the type of arguments must be compatible with an existing method in the class **X**,
2. the result type must match the expected type in the expression.

The overloading and the late binding mechanisms are the same as for the instance method invocations.

<i>General Information</i>	
Operator	<code>::</code>
Syntaxe	<code>identifier::identifier(expr_list)</code>
Type	n-ary
Operand Types	first operand: <b>identifier</b> , second operand: <b>identifier</b> , other operands: <i>any type</i>
Result Type	type of the atom returned by the method call
Functions	invokes the class method denoted by the second operand applied to the class denoted by the first operand, using the other operands as arguments. The first operand must denote an EYEDB class of an agregat including the class method whose name is the second operand. The number of arguments and the type of arguments must match one of the class methods included in the class denoted by the first operand.

<i>Expression Examples</i>		
<i>expression</i>	<i>result</i>	<i>comments</i>
<code>EyeDB::getVersion()</code>	2.8.0	<code>getVersion()</code> is a native static method of the class <code>EyeDB</code>
<code>Person::checkName("johnny")</code>	<i>the value returned by the method call</i>	the class method <code>checkName</code> must exist in the class <code>Person</code> and must take one and only one input string argument.

## 5.20 Eval/Unval Operators

### eval Operator

One major feature of OQL is that one can invoke its evaluator using the `eval` operator. This allows us to build OQL constructs at runtime and perform their evaluation. This is very useful, for instance, when we want to build a query expression where the projection or the `from` reference sets are unknown. For instance, the following function allows us to retrieve the values of the attribute *attrname* in the class *classname*:

```
function getValues(classname, attrname) {
  cmd := "select x." + attrname + " from " + classname + " x";
  return (eval cmd);
}
```

<i>General Information</i>	
Operator	<code>eval</code>
Syntax	<code>eval string</code>
Type	unary
Operand Types	<code>string</code>
Functions	calls the OQL evaluator on the string operand. The string operand can contain any OQL valid construct: an expression, a statement or a sequence of statements.

<i>Expression Examples</i>	
<i>expression</i>	<i>result</i>
<code>eval "10"</code>	10
<code>eval "a := 100"</code>	result is 100; the variable <code>a</code> is set to 100
<code>eval "a := \"hello\"; b := a + \"world\""</code>	result is "hello world"; the variable <code>a</code> is set to "hello"; the variable <code>b</code> is set to "hello world"

### unval Operator

The `unval` is the inverse of the `eval` in the sense that it takes any valid OQL expression and returns the string representation; the comments and, when not necessary, the spaces and tabulations are skipped. For instance, the construct `unval a := 10` returns `"(a:=10)"`.

<i>General Information</i>	
Operator	<code>unval</code>
Syntax	<code>unval expr</code>
Type	unary

Operand Types	<i>any type</i>
Functions	returns the string expression

<i>Expression Examples</i>	
<i>expression</i>	<i>result</i>
<code>unval 10</code>	"10"
<code>unval alpha += 10 - beta + 1</code>	"(alpha:=(alpha+((10-beta)+1)))"
<code>eval unval alpha := "hello"</code>	returns "hello"; alpha is set to "hello"

## 5.21 Set Expressions

OQL allows us to perform the following operations on **sets** and **bags**: union, intersection, difference and inclusion. The operands can be sets or bags. For all these operators, when the operand's collection types are different (**bag** and **set**), the set is first converted to a bag and the result is a bag.

### union Operator

The **union** operator performs the union of two **sets** or **bags**. This operator has the same precedence as the logical or operator.

<i>General Information</i>	
Operator	<b>union</b>
Syntax	<i>expr union expr</i>
Type	binary
Operand Types	<b>set</b> or <b>bag</b>
Result Type	<b>set</b> if both two operands are of type <b>set</b> , <b>bag</b> otherwise
Functions	returns the union of the two operands.

<i>Expression Examples</i>	
<i>expression</i>	<i>result</i>
<code>set(1, 2) union set(2, 3)</code>	<code>set(1, 2, 3)</code>
<code>set(1, 2) union bag(2, 3)</code>	<code>bag(1, 2, 2, 3)</code>
<code>list(1, 2) union bag(2, 3)</code>	<i>raises an error</i>

### intersect Operator

The **intersect** operator performs the intersection of two **sets** or **bags**. This operator has the same precedence as the logical and operator.

<i>General Information</i>	
Operator	<b>intersect</b>
Syntax	<i>expr intersect expr</i>
Type	binary
Operand Types	<b>set</b> or <b>bag</b>
Result Type	<b>set</b> if both two operands are of type <b>set</b> , <b>bag</b> otherwise
Functions	returns the intersection of the two operands.

<i>Expression Examples</i>
----------------------------

<i>expression</i>	<i>result</i>
<code>set(1, 2) intersect set(2, 3)</code>	<code>set(2)</code>
<code>set(1, 2) intersect bag(2, 3)</code>	<code>bag(2)</code>
<code>bag(1, 2, 2, 3) intersect bag(2, 3, 2)</code>	<code>bag(2, 2, 3)</code>
<code>list(1, 2) intersect bag(2, 3)</code>	<i>raises an error</i>

### except Operator

The **except** operator performs the difference between two **sets** or **bags**. This operator has the same precedence as the logical or operator.

<i>General Information</i>	
Operator	<b>except</b>
Syntax	<i>expr except expr</i>
Type	binary
Operand Types	<b>set</b> or <b>bag</b>
Result Type	<b>set</b> if both two operands are of type <b>set</b> , <b>bag</b> otherwise
Functions	returns the difference of the two operands.

<i>Expression Examples</i>	
<i>expression</i>	<i>result</i>
<code>set(1, 2) except set(2, 3)</code>	<code>set(1)</code>
<code>set(1, 2) except bag(2, 3)</code>	<code>bag(1)</code>
<code>set(1, 2, 10) except bag(12)</code>	<code>bag(1, 2, 10)</code>
<code>list(1, 2) except bag(2, 3)</code>	<i>raises an error</i>

### Inclusion Operators

The inclusion operators for **sets** and **bags** are the comparison operators *less than/greater than* introduced in a previous section.

<i>General Information</i>	
Operator	< <= > >=
Syntax	<i>expr &lt; expr</i> <i>expr &lt;= expr</i> <i>expr &gt; expr</i> <i>expr &gt;= expr</i>
Type	binary
Operand Types	<b>set</b> or <b>bag</b>
Result Type	<b>boolean</b>
Functions	<i>coll1 &lt; coll2</i> : returns <b>true</b> if and only if <i>coll1</i> is included in <i>coll2</i> but not equal to <i>coll2</i> <i>coll1 &gt; coll2</i> : returns <b>true</b> if and only if <i>coll2</i> is included in <i>coll1</i> and not equal to <i>coll1</i> <i>coll2 &lt;= coll1</i> : returns <b>true</b> if and only if <i>coll1</i> is included in <i>coll2</i> or equal to <i>coll2</i> <i>coll1 &gt;= coll2</i> : returns <b>true</b> if and only if <i>coll2</i> is included in <i>coll1</i> or equal to <i>coll1</i>

<i>Expression Examples</i>
----------------------------

<i>expression</i>	<i>result</i>
<code>set(1, 2) &lt; set(2, 3)</code>	<code>false</code>
<code>set(1, 2) &lt; set(2, 3, 1)</code>	<code>true</code>
<code>set(1, 2) &lt; bag(2, 3, 1)</code>	<code>true</code>
<code>set(1, 2) &lt;= bag(2, 1)</code>	<code>false</code>
<code>set(1, 2) &gt;= bag(2, 1)</code>	<code>false</code>

## 5.22 Object Creation

OQL allows us to create persistent or transient objects using the **new** operator. The general syntax for an object creation is as follows:

```
[new] [<[expr]>] class_name({path_expression : expr })
```

1. the operator **new** is optionnal: when the operator is missing, the construct is called an implicit **new** construct. When the optionnal following construct “<[expr]>” is not used, there is no functional differences between using **new** or not.
2. the optionnal construct after the **new** operator indicates the target location of the object to create:
  - (a) if this construct is omitted, the object will be a persistent object created in the current default database,
  - (b) if this construct is under the form <[expr]>, the OQL interpreter expects for a database object handle as the result of the expression evaluation. This database will be used as the target location. For instance:
 

```
new < oql$db > Person();
```

 will create a **Person** instance in the database pointed by **oql\$db**, which is in fact the current database.
  - (c) if this construct is under the form <>, the object will be a transient object.
3. the *class\_name* indicates the name of a valid user type in the context of the current database.
4. the *path\_expression* indicates an attribute name or a sequence of attributes using the optional array operator, for instance the following path expressions are valid for an object construction:

```
name
lastname
addr.street
addr.town[3]
spouse.name
```

5. the *expr* behind *path\_expression* is any OQL expression as soon as its result type matches the expected type of the *path\_expression*.
6. the order of evaluation of the expressions is in the {*path\_expression* : *expr*} sequence is from left to right.
7. the expression returns the oid of the created object.

For instance:

- **new Person()** creates a person with all its attributes uninitialized,
- **Person()** creates a person with all its attributes uninitialized,
- **new Person(name : "john")** creates a person with its attribute **name** initialized to **john**,
- **Person(name : "john")** creates a person with its attribute **name** initialized to **john**,
- **new Person(name : "john", age : 32, spouse : new Person(name : "mary"))** creates a person named **mary** and a person named **john**, age 32 whose **spouse** is the person **mary**.

The **new** operator can also be used to create basic type object. Note that in this case, the operator is mandatory. The syntax for basic type creation is as follows:

```
new [<[expr]>] basic_type (value).
```

1. where *basic\_type* denotes an ODL basic type. It may be one of the following type: **int32**, **int16**, **int32**, **char**, **byte**, **float** or **oid**. Note that the type **string** is not allowed here.



- the *value* must be an atomic value of an OQL type mapped from the ODL basic type

For instance:

- `new int(2)`
- `new float(2.3)`
- `new float(2)`
- `new char('a')`
- `new oid(first(select Person))`

Finally, the `new` operator can be also used to create collections. The syntax for collection creation is as follows:

`[new] [<[expr]>] coll_type< class_name [, coll_name]> ([collection of elements])`

- the `new` operator is optionnal,
- where *coll\_type* denotes type of the collection: `set`, `bag`, `array` or `list`,
- the *class\_name* denotes the name of the class of the elements of the collection, for instance `Person*`, `Car*`,
- coll\_name* is an optionnal string which denotes the name of the collection to create,
- the optionnal *collection of elements* within parenthesis contains the elements (generally oids) to insert initially in the created collection,
- the expression returns the oid of the created collection.

For instance:

- `new set<Person *>()` creates an empty `set` of persons,
- `new set<Person *>(list(select Person))` creates a `set` containing all the persons in the database,
- `new set<Person *, "all babies">(list(select Person.age < 1))` creates a `set` named `all babies` containing all the persons whose age is less than 1.
- `new array<Car *>()` creates an empty `array` of cars.
- `new array<int>(list(1, 2, 3, 4))` creates an `array` of integers initially containing 1, 2, 3, 4.
- `new set<int *> (list(new int(2), new int(10)))` creates a `set` of integer objects containing initially two integer objects.

General Information	
Operator	<code>new</code>
Syntax	<code>new [&lt;[<i>expr</i>]&gt;] class_name({<i>path_expression</i> : <i>expr</i> })</code>
Type	n-ary
Operand Types	<i>any type</i>
Result Type	<code>oid</code> or <code>object</code>
Functions	creates an persistent or transient object

Expression Examples	
<i>expression</i>	<i>result</i>
<pre>john := new Person(name: "john",                     lastname: "wayne",                     age : fib(10));</pre>	returns the oid of the created <code>Person</code> instance

<pre> new Person(name: "mary",             lastname: "poppins",             addr.town : "jungle",             addr.street[0] : 'a',             addr.street[1] : 'b',             spouse : john,             spouse.age : 72             ); </pre>	returns the oid of the created <b>Person</b> instance
--	---

### 5.23 Object Deletion

The **delete** unary operator is used to delete persistent objects.

<i>General Information</i>	
Operator	<b>delete</b>
Syntax	<b>delete</b> <i>expr</i>
Type	unary
Operand Types	<b>oid</b> or <b>object</b>
Result Type	the operand type
Functions	delete a transient or persistent object

<i>Expression Examples</i>	
<i>expression</i>	<i>result</i>
<b>delete first</b> (select Person)	the oid of the deleted <b>Person</b> instance
<b>delete new</b> Person()	the oid of the deleted <b>Person</b> instance
<b>for</b> (x in (select Person) <b>delete</b> x	the oids of the deleted <b>Person</b> instances

### 5.24 Collection Expressions

OQL introduces a few operators for object collection manipulation: one of them is the array deferencing operator “[ ]” (5.15) that is overloaded for ordered collection manipulation.

Some them are ODMG OQL compliant, the others are EYEDB extensions. Object collections may be persistent or transient, ordered or not. These operators allows us to make the following kind of operations:

1. gets the contents of a collection: operator **contents**,
2. get an element at a given position in an ordered collection: operator [ ] (ODMG compliant),
3. get elements at some given positions in an ordered collection: operators [ : ] (ODMG compliant) and [ ? ],
4. checks if an element is in a collection: operator **in** (ODMG compliant),
5. add an element in an unordered collection: operator **add/to**
6. suppress an element from an unordered collection: operator **suppress/from**,
7. set or suppress an element in an ordered at a given position: operator [ ],
8. set or suppress elements in an ordered at a given position: operators [ : ] and [ ? ],
9. append an element in an ordered collection: operator **append/to**,
10. checks if a given condition is realized for at least one element in a collection: operator **in** (ODMG compliant),
11. checks if a given condition is realized for all elements in a collection: operator **for/all** (ODMG compliant),

12. checks if a given condition is realized for a given number range of elements in a collection: operator extended **for**.

In all the following examples, the OQL variables **p0** denotes the first **Person** instance in the database: **p0 := first(select Person)**.

**Important Note:** although they are reference in the following descriptions of collection operators, the object collections **list** are not implemented in the current version of EYEDB.

#### contents Operator

The **contents** unary operator is used to give the contents of a given ordered or unordered object collection. It returned an OQL collection of the same type of the object collection: **set**, **bag**, **array** or **list**.

<i>General Information</i>	
Operator	<b>contents</b>
Syntax	<b>contents</b> <i>expr</i>
Type	unary
Operand Types	<b>oid</b> or <b>object</b> collection
Result Type	a collection of objects
Functions	returns the contents of an object collection

<i>Expression Examples</i>	
<b>contents(p0.children)</b>	an array of <b>Person</b> oids
<b>select contents(x.children) from Person x</b>	returns a list of arrays of <b>Person</b> oids.
<b>contents(list(1, 2, 3))</b>	raises an error: oid or object expected, got list

#### in Operator

The **in** operator is used to check if a given element is in ordered or unordered object collection.

<i>General Information</i>	
Operator	<b>in</b>
Syntax	<i>expr in expr</i>
Type	binary
Operand Types	first operand: <i>any type</i> , second operand: <b>oid</b> or <b>object</b> collection
Result Type	<b>boolean</b>
Functions	returns <b>true</b> if the first operand belongs to the collection pointed by the second operand; <b>false</b> otherwise

<i>Expression Examples</i>	
<b>first(select Person.name = NULL) in p0.children</b>	returns true if the first <b>Person</b> instance whose <b>name</b> is uninitialized is in the array of children of the first <b>Person</b>
<b>first(select Car.brand = "renault") in p0.cars</b>	returns true if the first <b>Car</b> instance whose <b>brand</b> equals <b>renault</b> is in the set of cars of the first <b>Person</b>

#### add/to Operator

The **add/to** operator is used to add an element in an unordered collection (**set** or **bag**).

<i>General Information</i>	
Operator	<b>add/to</b>
Syntaxes	<b>add</b> <i>expr</i> <b>to</b> <i>expr</i>
Type	binary
Operand Types	first operand: <i>any type</i> , second operand: <b>oid</b> or <b>object</b> unordered collection ( <b>set</b> or <b>bag</b> )
Result Type	type of the first operand
Functions	adds the first operand to the non-indexed collection (i.e. bag or set) pointed by the second operand; returns the first operand.

<i>Expression Examples</i>	
<b>add new Car</b> (num : 100) <b>to</b> p0.cars	returns the created <b>Car</b> oid
<b>add new Person</b> (name : "john") <b>to</b> p0.children	<i>raises an error: cannot used non indexed insertion in an array</i>
<b>add new Car</b> () <b>to new set</b> <Car *>()	returns the just created car; but we have <i>lost</i> the oid of the just created <b>set</b> of cars!
<b>add new Person</b> () <b>to</b> (c := <b>new bag</b> <Person *>())	returns the just created person; the created bag has been kept in the OQL variable c

## [] Operator

The polymorphic [] operator is used to set or get an element in an ordered collection (**array** or **list**) at a given position: it can be used in a right or left value.

<i>General Information</i>	
Operator	<b>[]</b>
Syntaxes	<i>expr</i> [ <i>expr</i> ]
Type	binary
Operand Types	first operand: collection <b>array</b> or <b>list</b> , second operand: <b>integer</b>
Result Type	the type of the element,
Functions	gets the element in the collection pointed by the first operand at the position pointed by the second operand. If used at a left value, gets a reference to that element.

<i>Expression Examples</i>	
p0.children[0]	returns the child at position #0 in p0.children collection. Returns <b>nil</b> if there is no child at this position
p0.children[0] := <b>Person</b> (name : "john")	returns the created <b>Person</b> oid
p0.children[12039] := <b>Person</b> (name : "henry")	returns the created <b>Person</b> oid. This expression is valid in any case as the collection <b>arrays</b> automatically increased its size
( <b>array</b> < <b>Person</b> *>())[0] := <b>new Car</b> (num : 100)	returns the just created person; but the created array has been " <i>lost</i> " as it is not tied to any instance and as we did not bind it to any OQL variable

<code>(c := array&lt;Person *&gt;())[0] := new Car(num : 100)</code>	returns the just created person; the created array has been kept in the OQL variable <code>c</code>
<code>p0.cars[1] := Car(num : 100)</code>	<i>raises an error: array expected, got set</i>

### [:] Operator

The polymorphic `[:]` operator is used to set or get some elements in an ordered collection (**array** or **list**) at some given positions: it can be used in a right or left value. When used in a right value, the returned atom is a **set** of **struct** with the two attributes **index** and **value**. In each **struct** element returned, the value of **index** is the position of the element, the value of **value** is the element value. Note the returned **struct** elements are not ordered according to the element positions; it is why a **set** is returned. When used as a left value, the returned atom is a **set** of references on the elements .

General Information	
Operator	<code>[:]</code>
Syntaxes	<code>expr [ expr : expr ]</code>
Type	ternary
Operand Types	first operand: collection <b>array</b> or <b>list</b> , second and third operands: <b>integer</b>
Result Type	the type of the element,
Functions	gets the elements in the collection pointed by the first operand at the position range pointed by the second and third operands. If used at a left value, gets references to that elements.

Expression Examples	
<code>p0.children[0:1]</code>	returns a set of <b>struct</b> including the children and the position of the children position #0 and #1 in the <code>p0.children</code> collection. For instance: <code>set(struct(index : 0, value : 3874.33.293847:oid), struct(index : 1, value : 2938.33.1928394:oid))</code>
Returns <code>nil</code> if there is no child at these positions	
<code>p0.children[0:4] := Person(name : "john")</code>	Sets all the children at the position #0 to #4 to a new <b>Person</b> instance.
returns the created <b>Person</b> oid	
<code>p0.children[12000:12039] := Person(name : "henry")</code>	returns the created <b>Person</b> oid. This expression is valid in any case as the collection <b>arrays</b> automatically increased its size
<code>(array&lt;Person *&gt;(list(Person())))[0]</code>	returns the just created person within the just created array. But the array is “lost” as it is not tied to any instance and as we did not bind it to any OQL variable
<code>(x := array&lt;Person *&gt;(list(Person())))[0]</code>	returns the just created person; the created array has been kept in the OQL variable <code>c</code>

### [?] Operator

The polymorphic `[?]` operator is used to set or get all the elements in an ordered collection (**array** or **list**). It can be used in a right or left value. When used in a right value, the returned atom is a **set** of **struct** with the two attributes **index** and **value**. In each **struct** element returned, the value of **index** is the position of the element, the value of **value** is the element value. Note the returned **struct** elements are not ordered according to the element positions; it is why a

**set** is returned.

When used as a left value, the returned atom is a **set** of references on the elements .

General Information	
Operator	[?]
Syntaxes	<i>expr</i> [?]
Type	unary
Operand Type	collection <b>array</b> or <b>list</b> ,
Result Type	a set of <b>struct</b> or a set of references
Functions	gets all the elements in the collection pointed by the first operand If used at a left value, gets references to that elements.

Expression Examples	
<code>p0.children[?]</code>	returns a set of <b>struct</b> including the children and the position of all the children in the <code>p0.children</code> collection. For instance: <code>set(struct(index : 0, value : 3874.33.293847:oid), struct(index : 1, value : 2938.33.1928394:oid))</code>
Returns <b>nil</b> if the collection is empty	
<code>p0.children[?] := Person(name : "john")</code>	Sets all the children to a new <b>Person</b> instance.
returns the created <b>Person</b> oid	
<code>(array&lt;Person *&gt;(list(Person(), Person())))[?]</code>	returns a <b>set</b> of <b>struct</b> including the just created <b>Person</b> instances in the just created array.

### append/to Operator

The **append/to** operator is used to append an element to an ordered collection (**list** or **array**).

General Information	
Operator	<b>append</b>
Syntaxes	<b>append</b> <i>expr</i> to <i>expr</i>
Type	binary
Operand Types	first operand: <i>any type</i> , second operand: <b>oid</b> or <b>object</b> denoting an ordered collection
Result Type	<i>any type</i>
Functions	appends the element denoted by the first operand to the indexed collection (i.e. list or array) denoted by the second operand.

Expression Examples	
<code>append Person() to p0.children</code>	the created <b>Person</b> instance
<code>append Car() to p0.cars</code>	raises an error: <b>array</b> or <b>list</b> expected, got <b>set&lt;Person*&gt;</b>

### suppress/from Operator

The **suppress/from** operator is used to suppress an element from an ordered collection (**set** or **bag**).

<i>General Information</i>	
Operator	<b>suppress/from</b>
Syntaxes	<b>suppress <i>expr</i> from <i>expr</i></b>
Type	binary
Operand Types	first operand: <i>any type</i> , second operand: <b>oid</b> or <b>object</b> collection
Result Type	type of the first operand
Functions	suppress the first operand from the non-indexed collection (i.e. bag or set) pointed by the second operand; returns the first operand.

<i>Expression Examples</i>	
<b>suppress (select Car.num = 1000) from p0.cars</b>	the suppressed car if it was found in the collection; otherwise, <i>raises an error</i>
<b>suppress new Car() from p.cars</b>	<i>raises an error: item '71238.13.3959935:oid' not found in collection</i>
<b>suppress p0 from p0.children</b>	<i>raises an error: cannot used non indexed suppression in an array</i>

### empty Operator

The **empty** operator is used to empty an ordered or an unordered collection.

<i>General Information</i>	
Operator	<b>empty</b>
Syntax	<b>empty <i>expr</i></b>
Type	unary
Operand Types	<b>oid</b> or <b>object</b> collection
Result Type	<b>nil</b>
Functions	empty the collection pointed by the operand

<i>Expression Examples</i>	
<b>empty(first (select Person).children)</b>	<b>nil</b>
<b>empty(first (select Person).cars)</b>	<b>nil</b>
<b>empty new set&lt;Car *&gt;(list(new Car()))</b>	<b>nil</b> ; this expression creates a collection of <b>Car</b> containing initially a new <b>Car</b> , and empty it!

### in Operator

The **in** operator is used to check if a given condition is realized for at least one element in an ordered or unordered collection.

<i>General Information</i>	
Operator	<b>in</b>
Syntax	<b><i>identifier</i> in <i>expr</i> : <i>expr</i></b>
Type	ternary
Operand Types	first operand: <b>identifier</b> , second operand: <b>oid</b> or <b>object</b> collection, third operand: <b>boolean</b>
Result Type	<b>boolean</b>

Functions	returns <b>true</b> if it exists in the collection pointed by the second operand an element for which the third operand is evaluated to <b>true</b> .
-----------	---

<i>Expression Examples</i>	
<code>x in p0.children: x.name = "mary"</code>	<b>true</b> or <b>false</b>
<code>x in p0.cars: x.num &lt; 100 and x.num &gt;= 90</code>	<b>true</b> or <b>false</b>

### for Operator

The **for/all** operator is used to check if a given condition is realized for all elements in an ordered or unordered collection. This operator is ODMG compliant.

<i>General Information</i>	
Operator	<b>for/all</b>
Syntaxes	<b>for all</b> <i>identifier</i> <b>in</b> <i>expr</i> : <i>expr</i>
Type	ternary
Operand Types	first operand: <b>identifier</b> , second operand: <b>oid</b> or <b>object</b> collection, third operand: <b>boolean</b>
Result Type	<b>boolean</b>
Functions	returns <b>true</b> if for all items contained in the collection pointed by the second operand the third operand is evaluated to <b>true</b> .

<i>Expression Examples</i>	
<code>for all x in p0.children: x.name == "john"</code>	<b>true</b> or <b>false</b>
<code>for all x in p0.cars: x.num % 10</code>	<b>true</b> or <b>false</b>

The **for/cardinality** operator is used to check if a given condition is realized for a given number range of elements in an orderer or unordered collection. Note that this operator is the generalisation of the **in** and **for/all** operators:

**for** <0:\$> is equivalent to **in**

**for** <\$> is equivalent to **for all**

<i>General Information</i>	
Operator	<b>forcardinality</b>
Syntaxes	<b>for</b> < <i>expr</i> : <i>expr</i> > <i>identifier</i> <b>in</b> <i>expr</i> : <i>expr</i> <b>for</b> < <i>expr</i> > <i>identifier</i> <b>in</b> <i>expr</i> : <i>expr</i>
Type	5-ary
Operand Types	first and optional second operands: <b>integer</b> or <b>\$</b> , where <b>\$</b> denotes the collection cardinality, third operand: <b>oid</b> or <b>object</b> , fourth operand: <b>identifier</b> , fifth operand: <b>boolean</b>
Result Type	<b>boolean</b>
Functions	returns <b>true</b> if the number of items in the collection pointed by the third operand for which the third operand is evaluated to <b>true</b> is in the interval [first operand, second operand].



<i>Expression Examples</i>	
for <0:4> x in p0.children: x.name == "john"	true if at most 4 children have their name equals to john
for <4> x in p0.children: x.name == "john"	true if one an only one children have its name equals to john
for <0:\$> x in p0.cars: x.num = 10	equivalent to in
for <\$> x in p0.cars: x.num = 10	equivalent to for/all

## 5.25 Exception Expressions

Currently, EYEDB OQL does not provide full support for exception management as there is no **try/catch** operator. Nevertheless, the **throw** operator allows us to raise an error message, for instance:

```
if (!check(p))
  throw "variable p is not correct".
```

The **throw** operator stops the current thread of statements and returns the error message at the uppest level. In the following code:

```
a := 1;
throw "this is an error";
b := 2;
```

the variable **a** will be assigned to 1, but the variable **b** will not be assigned to 2 as the **throw** expression deroutes the normal thread of statements. The **throw** operator is often used in the body of functions,

<i>General Information</i>	
Operator	<b>throw</b>
Syntax	<b>throw</b> <i>expr</i>
Type	unary
Operand Type	<b>string</b>
Result Type	<b>nil</b>
Functions	raises an error message

<i>Expression Examples</i>	
throw "error message"	<b>nil</b>
throw "error #1: " + msg	<b>nil</b>

## 5.26 Function Definition Expressions

As introduced previously, OQL supports functions. There are two types of functions definition syntax: function definition expression and function definition statements. The first ones, exposed in this section, are more restrictive than the second ones, as their definition can contain only one expression. The second ones contain an unbounded sequence of statements. The function definition expressions are ODMG compatible and are called *Named Query Definition* in the ODMG standard. To define such a function, one must use the operator **define/as**. The general syntax for a definition function expression is:

```
define identifier [(arglist)] as expr.
```

For instance:

```
define div2(x) as x/2;
div2(10); // returns 5
```

```
define pimul(x) as x * 3.1415926535;
pimul(23); // returns 72.256631
```

```
define getOnePerson(name) as first(select Person.name = name);
getOnePerson("john"); // returns an oid or nil
```

As the last operand of the **define/as** operator is any OQL expression, it can be a sequence of expressions by using the comma sequencing operator. Therefore, the following construct is valid:

```
define getit(age) as ::getit_call++,
                    list_of_persons := select Person.age >= age,
                    (count(list_of_persons) > 0 ? list_of_persons[0] : nil);

getit(10); // returns the first Person whose age is greater or equal to
           10, or nil
getit_call; // equals 1
list_of_persons; // raises an error: uninitialized identifier 'list_of_persons'

getit(20);
getit_call; // equals 2
```

Several comments about this code:

1. `::getit_call` denotes a global variable, while `list_of_persons` denotes a variable local to the function: the variable scoping has previously been introduced in the identifier expressions, and will be explained again in the function definition statements Section.
2. as there are no iteration expression (for instance a **for** or **while** expression) and as a function definition expression can contain only one expression, one cannot use a function definition expression with iteration statements. To use an iteration statement, one needs to use a function definition statement.
3. when one needs to define a function with a sequence of expressions, it may be easier and more readable to use a function definition statement instead of a function definition expression. The *statement* version of the previous function is:

```
function getit(age) {
  ::getit_call++;
  list_of_persons := select Person.age >= age;
  if (count(list_of_persons))
    return list_of_persons[0];
}
```

which is - for C, C++ or Java programmers - a more natural construct.

Finally, the function definition allows us for recursive definitions, for instance:

```
define fib(n) as (n < 2 ? n : fib(n-2) + fib(n-1));
fib(10); // returns 55
```

General Information	
Operator	<b>define/as</b>
Syntax	<b>define</b> <i>identifier</i> [ <i>arglist</i> ] <b>as</b> <i>expr</i>
Type	n-ary
Operand Type	first operand: <b>identifier</b> , last operand: <i>any type</i> , other operands: <b>identifier</b>
Result Type	<b>identifier</b>
Functions	defines a function

Expression Examples	
<b>define</b> Persons <b>as</b> select Person	Persons
<b>define</b> fact(n) <b>as</b> (n < 2 ? n : n * fact(n-1))	fact

## 5.27 Conversion Expressions

The conversion unary operators **string**, **int**, **char**, **float**, **oid** and **ident** allows us to make the following conversions:

<i>Operator</i>	<i>From</i>	<i>To</i>	<i>Returned Atom</i>
<b>string</b>	<i>any type</i>	<b>string</b>	the string representation of the operand
<b>int</b>	<b>int</b>	<b>int</b>	the <b>int</b> operand
	<b>char</b>	<b>int</b>	the operand casted to an <b>int</b>
	<b>float</b>	<b>int</b>	the operand casted to an <b>int</b>
	<b>string</b>	<b>int</b>	the operand converted to an <b>int</b>
<b>char</b>	<b>char</b>	<b>char</b>	the <b>char</b> operand
	<b>int</b>	<b>char</b>	the operand casted to a <b>char</b>
	<b>float</b>	textttchar	the operand casted to a <b>char</b>
	<b>string</b>	<b>char</b>	the operand converted to a <b>char</b>
<b>float</b>	<b>float</b>	<b>float</b>	the <b>float</b> operand
	<b>char</b>	<b>float</b>	the operand casted to a <b>float</b>
	<b>int</b>	<b>float</b>	the operand casted to a <b>float</b>
	<b>string</b>	<b>float</b>	the operand converted to a <b>float</b>
<b>oid</b>	<b>oid</b>	<b>oid</b>	the <b>oid</b> operand
	<b>string</b>	<b>oid</b>	the <b>string</b> operand converted to an <b>oid</b>
<b>ident</b>	<b>ident</b>	<b>ident</b>	the <b>ident</b> operand
	<b>string</b>	<b>ident</b>	the <b>string</b> operand converted to an <b>ident</b>

These operators are used to perform an explicit conversion such as convert the string "123" to the integer 123, or to perform an explicit cast for numbers such as casting the integer 10 to the float 10.0. These operators evaluate first their operand before performing the conversion. If the operand type is valid, no error is raised even if its format is not valid, for instance: **int** "alpha" returns 0, while **oid** "aoaoai" returns NULL. Note that because of the precedence of these operators, parenthesis are necessary to make a conversion of a non-primary operand. For instance, **string** 1+2 is not valid: you should use **string** (1+2).

### string operator

The **string** operator evaluates its operand and returns its string representation.

<i>General Information</i>	
Operator	<b>string</b>
Syntax	<b>string</b> <i>expr</i>
Type	unary
Operand Type	any type
Result Type	<b>string</b>
Function	returns the string representation of any atom

<i>Expression Examples</i>	
<b>string</b> 123.3	"1203.300000"
<b>string</b> 'a'	"a"
<b>string</b> first(select Person)	"71211.13.1486847:oid"
<b>string</b> &alpha	"::alpha"
<b>string</b> list(1, 2, 3+2)	"list(1, 2, 5)"
<b>string</b> (list("hello", 30) + list(10))	"list("hello", 30, 10)"
<b>string</b> (1+3)	"4"
<b>string</b> 1+3	<i>raises an error</i>

### int operator

The **int** operator evaluates its operand and converts or casts it to an integer.

If the operand is the string, it converts it using the **atoi** C function. If the string is not a valid integer, it returns a 0.

If the operand is a char or float, it casts it to an integer.

If the operand is an integer, it returns it.

<i>General Information</i>	
Operator	<b>int</b>
Syntax	<b>int</b> <i>expr</i>
Type	unary
Operand Type	<b>int</b> , <b>char</b> , <b>float</b> or <b>string</b>
Result Type	<b>int</b>
Function	returns the integer conversion or cast of the operand

<i>Expression Examples</i>	
<b>int</b> 123.3	123
<b>int</b> 12	12
<b>int</b> 'a'	97
<b>int</b> "123"	123
<b>int</b> ("123" + "12")	12312
<b>int</b> alpha	the value of <b>alpha</b> converted or casted to an integer
<b>int</b> list(1, 2, 3)	raises an error

### char operator

The **char** operator evaluates its operand and converts or casts it to a **char**.

If the operand is the string of length one, it returns the character of this string. If the string has several characters, it returns a '\000'.

If the operand is a integer or float, it casts it to a character.

If the operand is a character integer, it returns it.

<i>General Information</i>	
Operator	<b>char</b>
Syntax	<b>char</b> <i>expr</i>
Type	unary
Operand Type	<b>int</b> , <b>char</b> , <b>float</b> or <b>string</b>
Result Type	<b>char</b>
Function	returns the character conversion or cast of the operand

<i>Expression Examples</i>	
<b>char</b> 123.3	{
<b>char</b> 'a'	'a'
<b>char</b> alpha	the value of <b>alpha</b> converted or casted to a character
<b>char</b> "a"	'a'
<b>char</b> "hello"	'^@'
<b>char</b> list(1, 2, 3)	raises an error

### float operator

The **float** operator evaluates its operand and converts or casts it to a float.

If the operand is the string, it converts it using the **atof** C function. If the string is not a valid float, it returns a 0.0.

If the operand is a integer or float, it casts it to a float.

If the operand is a float, it returns it.

<i>General Information</i>	
Operator	<b>float</b>

Syntax	<b>float</b> <i>expr</i>
Type	unary
Operand Type	<b>int</b> , <b>char</b> , <b>float</b> or <b>string</b>
Result Type	<b>float</b>
Function	returns the float conversion or cast of the operand

<i>Expression Examples</i>	
<b>float</b> 123.0	123.0
<b>float</b> 123.3	123.3
<b>float</b> 'a'	97.000
<b>float</b> "123.0000000"	123.0
<b>float</b> ("123." + "12")	123.12
<b>float</b> "hello"	0.0
<b>float</b> alpha	the value of <b>alpha</b> converted or casted to a float
<b>float</b> list(1, 2, 3)	raises an error

**oid operator**

The **oid** operator evaluates its string operand and returns the corresponding oid. If the string does not denote a valid oid, the NULL oid is returned.

If the operand is an oid, it returns it.

<i>General Information</i>	
Operator	<b>oid</b>
Syntax	<b>oid</b> <i>expr</i>
Type	unary
Operand Type	<b>oid</b> or <b>string</b>
Result Type	<b>oid</b>
Function	returns the oid denoted by the string operand

<i>Expression Examples</i>	
<b>oid</b> "234.34.33:oid"	234.34.33:oid
<b>oid</b> 234.34.33:oid	234.34.33:oid
<b>oid</b> first(select Person)	returns the first person oid
<b>oid</b> 'a'	raises an error

**ident operator**

The **ident** operator evaluates its string operand and returns the corresponding identifier. If the operand is an identifier, it returns it.

<i>General Information</i>	
Operator	<b>ident</b>
Syntax	<b>ident</b> <i>expr</i>
Type	unary
Operand Type	<b>ident</b> or <b>string</b>
Result Type	<b>string</b>
Function	returns the identifier denoted by the string operand

<i>Expression Examples</i>	
<code>ident "alpha"</code>	<code>alpha</code>
<code>ident "alpha#1x"</code>	<code>alpha#1x</code>
<code>ident "alpha" := 123</code>	123, alpha has been assigned to 123
<code>valof &amp;(ident "alpha")</code>	123
<code>ident 'a'</code>	raises an error

## 5.28 Type Information Expressions

OQL provides two type information unary operators: `typeof` and `classof`. The first one takes any operand type, while the second one takes an oid or an object operand. Note that because of the precedence of these operators, parenthesis are necessary to get type information about a non-primary operand. For instance, `typeof 1+2` is not valid: you should use `typeof (1+2)`.

### `typeof` operator

The `typeof` operator is used to get the type of any OQL atom. It evaluates its operand and returns the string type of its operand. For instance: `typeof 1` returns `"int"` while `typeof "hello"` returns `"string"`.

<i>General Information</i>	
Operator	<code>typeof</code>
Syntax	<code>typeof expr</code>
Type	unary
Operand Type	any type
Result Type	<code>string</code>
Function	returns the type of the atom

<i>Expression Examples</i>	
<code>typeof "alpha"</code>	<code>"string"</code>
<code>typeof (1+20.)</code>	<code>"float"</code>
<code>typeof list(1, 2, 3)</code>	<code>"list"</code>
<code>typeof first(select Person)</code>	<code>"oid"</code>
<code>typeof 1+3049</code>	raises an error
<code>typeof alpha</code>	type of the value of <code>alpha</code>
<code>typeof &amp;alpha</code>	<code>ident</code>

### `classof` operator

### `typeof` operator

The `classof` operator is used to get the class of any oid or object. It evaluates its operand and returns the string class of its operand. For instance: `classof first(select Person)` returns `"Person"` while `typeof new Car()` returns `"Car"`.

<i>General Information</i>	
Operator	<code>classof</code>
Syntax	<code>classof expr</code>
Type	unary
Operand Type	oid or object
Result Type	<code>string</code>
Function	returns the class of the operand

<i>Expression Examples</i>
----------------------------

<code>classof first(select class)</code>	"basic_class"
<code>classof (c := new Car(num : 10))</code>	"Car"
<code>classof NULL</code>	" "
<code>classof first(select Person).spouse</code>	"Person" or NULL
<code>classof 1</code>	raises an error

## 5.29 Query Expressions

The **select** operator is used to perform queries in a database. The general syntax of this operator is as follows:

```
select [distinct] projection [from fromList [where predicat] [order by orderExprList [asc|desc]]
```

1. **distinct** means that duplicates must be eliminated,
2. *projection* is an expression using the variables defined in the *fromList*,
3. *fromList* is a sequence of comma-separated items under one of the following forms:  

```
var in expr
expr as var
expr var
```

where *expr* is an expression of type **collection** or is a class name, and *var* is the name of a variable.
4. *predicat* is a boolean expression using the variables defined in the *fromList*,
5. *orderExprList* is a comma-separated list of sortable expressions (i.e. atomic type).
6. **asc** means that the order should be performed in an ascendant way (the default) and **desc** means the inverse.

### ODMG vs. EyeDB OQL Query Expressions

As explained in the Section *OQL vs. ODMG 3 OQL*, there are a few differences between ODMG OQL and EYEDB OQL query expressions:

- the **having/group** clause is not supported in the current implementation.
- in the current implementation, one cannot use implicit **from** clause (i.e. **from** clause without variables). ODMG OQL supports constructs such as: **from Person** without any variable. This implementation does not.
- contrary to ODMG OQL, the **from** clause is optionnal, A **select** expression which does not use the **from** is called an implicit **select** expression.
- in a **from** clause such as “**x in expr**”, “**expr**” can be the name of a class. In this case, the interpreter understand this as the extent of this class. ODMG OQL does not support that.
- the SQL specific aggregate operators **min(\*)**, **max(\*)**, **count(\*)**, **sum(\*)** and **avg(\*)** are not supported
- the **order by** clause is more restrictive than in the ODMG OQL specifications (see below).
- the **select \*** is not supported in the current implementation.

### The general select syntax

Rather than introducing the **select** operator in a formal way by using a lot mathematical symbols, we introduce it first, in an unformal way, and then, through query examples.

The unformal description of the query process is as follows:

1. The **from** clause determine the sets of objects on which the query will be applied. For instance, in the **from** clause, “**x in Person, y in x.children**”, the sets on which the query will be applied are all the **Person** instances bound to the variable **x** and the children of these instances, bound to the variable **y**.
2. These sets of objects are filtered by retaining only the objects that satisfy the **predicat** in the **where** clause. These result objects are gathered into a **bag**. If no **where** clause is there, all objects are retained.
3. If an **order by** clause is present, a sort is performed using to the following process:
  - (a) each order expression must be of a sortable type: number (**int**, **char** or **float**) or **string**. If not, an error is raised.

- (b) the **bag** is ordered into a **list** according to the first order expression. Then identical atoms are ordered again using the second order expression and so on.
  - (c) there is a restriction in the current implementation: each expression in the *orderExprList* must be present in the *projection* expression. If not present, an error is raised.
4. The *projection* expression is evaluated for each object in the **collection** and the results of these evaluations are gathered into a **bag**.
  5. If the keyword **distinct** is there, the eventual duplicates are eliminated.
  6. Finally, if the **order by** clause is present, the result **bag** is converted to a **list**; if the **distinct** keyword is there without an **order by**, the **bag** is converted to a **set**; and if neither **order by** nor **distinct** are used, we get a **bag**.

The following table presents several examples:

<i>Simple select/from Examples</i>	
<b>select x from Person x</b>	returns a <b>bag</b> containing all the <b>Person</b> oids in the database
<b>select x.name from Person x</b>	returns a <b>bag</b> containing the name of every <b>Person</b> instance in the database
<b>select struct(name: x.name, age: x.age) from Person x</b>	returns a <b>bag</b> containing <b>struct</b> elements including the name and age of every <b>Person</b> instance in the database
<b>select list(x, x.name) from Person x where x.name ~ "h"</b>	returns a <b>bag</b> of <b>list</b> elements containing the oid and the name of every <b>Person</b> instances whose name matches the regular expression "h"
<b>select list(x, x.name) from Person x where x.name ~ "h" order by x.name</b>	same as previous example, but the result is a <b>list</b> ordered by the name of the persons
<b>select x from Person x where x.spouse.name = "john" or x.age &lt; 10</b>	returns a bag of <b>Person</b> instances whose spouse name is equal to "john" or the age is less than 10
<b>select x from Person x order by x.name</b>	current implementation restriction: <i>raises an error: x.name not found in projection</i>

### Arrays and collections in query expressions

OQL provide supports for performing direct queries through non-collection array and collection attributes without explicit joins. To perform such queries, one must use the operators `[]`, `[?]` or `[:]`. All the operators may be used for non-collection arrays. For collections (**list**, **set**, **bag** and **array**), only the operator `[?]` is valid.

The operator `[]` denotes an element in a non-collection array.

The operator `[:]` denotes a range of elements in a non-collection array.

The operator `[?]` denotes all elements in a non-collection array or in a collection. The following table presents several examples:

<i>Array and Collection based Query Examples</i>	
<b>select x.name from Person x where x.name[0] = 'j'</b>	returns all the <b>Person</b> instances whose name begins with a 'j'
<b>select x from Person x where x.other_addrs[0].street ~~ "par."</b>	returns all the <b>Person</b> instances whose first <b>other_addrs</b> street matches the regular expression "par."
<b>select x from Person x where x.other_addrs[?].street ~~ "par.."</b>	returns all the <b>Person</b> instances whose any <b>other_addrs</b> street matches the regular expression "par.."
<b>select x from Person x where x.other_addrs[1:3].street ~~ "par.."</b>	returns all the <b>Person</b> instances whose the first, second or third <b>other_addrs</b> street matches the regular expression "par.."
<b>select x from Person x where x.children[?].name = "johnny"</b>	



returns all the persons whose one of its children is called "johnny"
<code>select x from Person x where x.cars[?].num &lt; 100 or x.children[?].name = "mary"</code>
returns all the persons whose one of its cars has a number less than 100 or a child called "mary"
<code>select x from Person x where x.children[1].name = "johnny"</code>
although the <code>children</code> is a collection array, an error is raised. This is a current limitation of the implementation that will disappear soon

### The Implicit select syntax

An implicit **select** expression is a **select** expression with neither a **from** nor an explicit **where** clause. In fact, the **where** clause may be included in the *projection* expression.

This particular syntax has the advantage to be more compact and more simple than the general syntax, but some queries cannot be performed:

1. queries performing an explicit join,
2. queries having **or** or **and** in their **where** clause.

The following table presents several examples:

<i>Simple Implicit select Examples</i>
<code>select 1</code> returns 1
<code>select Person</code> returns a <b>bag</b> containing all <b>Person</b> instances in the database
<code>select Person.name</code> returns a <b>bag</b> containing the name of every <b>Person</b> instances in the database
<code>select Person.name = "john"</code> returns a <b>bag</b> containing the oids of every <b>Person</b> instances whose name is equal to "john"
<code>(select distinct Person.name = "john").age</code> returns a <b>set</b> containing the age of every <b>Person</b> instances whose name is equal to "john"
<code>select Person.name = "john" or Person.age = 10</code> <i>raises an error: use select/from/where clause</i>
<code>select Person.name order by Person.name</code> returns a <b>list</b> containing of the sorted names of all <b>Person</b> instances

### Querying the schema

As every abstraction is an object, one can perform queries on the schema and classes. For instance, to get the oids of all the existing classes in a schema:

`select schema` which is equivalent to `select class`.

In the EYEDB object model, the class `class` has the following native attributes (some are inherited from the `object` class):

1. **class** (inherited from `object`) is the class of this class,
2. **protection** (inherited from `object`) is the protection object of this class,
3. **type** is the type of the class: "system" or "user".
4. **name** is the name of this class,
5. **parent** is the parent class of this class,
6. **extent** is the extent collection of this class: contains all the object instances of this class,
7. **components** is the collection of components of this class: contains the constraints, the methods, the triggers, the index.

Queries can be performed according to one or more of the previous attributes.

<i>Query Schema Examples</i>	
<code>select class.name = "Person"</code>	returns a <b>bag</b> containing the class whose name is "Person"
<code>select class.type = "user"</code>	returns all the user classes
<code>select x from class x where x.name ~ "P" and x.type = "user"</code>	returns the user classes whose name matches the given regular expression
<code>select class.parent.name = "Person"</code>	returns a <b>bag</b> containing the sub-classes of the class <b>Person</b>

### How queries are optimized?

EYEDB queries implementation make an heavy use of index. Index may be used as terminal index or as non-terminal index: terminal index are those used at the end of a path expression, and the other are those used inside a path expression.

For instance, assuming that each attribute in our schema is indexed, the query `select Person.name = "john"` uses the index on the attribute **name** as a terminal index.

The query `select Person.spouse.age < 23` uses the index on the attribute **age** as a terminal index, and the index on the indirect attribute **spouse** as a non-terminal index, while the query `select Person.spouse = 6252.3.48474:oid` uses the index on the **spouse** attribute as a terminal index.

The query `select Person.children[?].spouse.name = "mary"` uses the index of the attributes **name**, **spouse** and of the literal collection attribute **children**.

The queries with a **where** clause containing logical **and** constructs are optimized so to take advantage of the index. For instance, assuming that there is an index on the **name** attribute and no index on the **age** attribute, the query `select x from Person x where x.age = 100 and x.name = "john"` will perform first the query on the **name** attribute and then filter the result according to the given predicat on the **age** attribute.

If the two expressions around the logical **and** operator have an index or if none of the two expression has a index, the interpreter performs first the left part and then the second part. So, in this case, the order of the two expressions around the **and** is important.

Finally, the query `select Person.name` reads directly the index of the **name** attribute, instead of reading the name of each **Person** instance.

Nevertheless, currently, there are some constructs that the OQL interpreter does not interpret cleverly and where index are not used. This is unfortunately the case of the join constructs such as:

```
select x from Person x, x.spouse y where y.name = "mary".
```

This construct will not make use of the index on the **spouse** attribute (but it will use the index of the **name** attribute). Fortunately, in a lot of cases, join queries may be replaced by a path expression query, for instance:

```
select x from Person x where x.spouse.name = "mary" is the alternate form the previous join query.
```

This alternate form (path-expression oriented) is the preferred one, because:

- it is more intuitive,
- it is more *object oriented* (please forget relationnal!),
- it is more compact,
- it uses index properly.

Of course, the last reason is not a good reason as a proper query implementation should use index whatever the used syntax. The implementation will be improved in a next version.

Note that query optimisations are fragile and do not believe that the OQL interpreter knows your data better than you. So, the two following simple rules should be applied:

1. when path expression oriented queries are enough, do not use join constructs,
2. in a **and** expression within a **where** clause, put the expression which denotes the most little set of instances on the left side of the **and**. In some particular cases, the OQL interpreter knows how to optimize the **and**, but in most of the cases, it does not.

Some experimental hints can be added to an **and** expression in the **where** clause to help the interpreter to do the things better, but there are currently too much experimental to be documented here.

### 5.30 Miscellenaous Expressions

A few OQL operators cannot be easily classified in one of the previous categories. We choose to classify them in the miscellenaous operators. These operators are: **bodyof**, **structof**, **[!]** and **import**.

#### **bodyof operator**

The **bodyof** operator is used to get the body of an OQL function. For instance, let the function **fib**: **define fib(n) as (n < 2 ? n : fib(n-2) + fib(n-1))**. The expression **bodyof fib** will return: **"fib(n) ((n<2)?n:(fib((n-2))+fib((n-1))))"**. This operator could be applied to expression-functions (i.e. functions defined with the **define/as** operator) or statement-functions (i.e. functions defined with the **function** operator).

<i>General Information</i>	
Operator	<b>bodyof</b>
Syntax	<b>classof</b> <i>expr</i>
Type	unary
Operand Type	ident denoting a function
Result Type	<b>string</b>
Function	returns the body of the function

<i>Expression Examples</i>	
<b>bodyof is_int</b>	<b>"is_int(x) ((typeof x)=="integer")"</b>
<b>bodyof first</b>	<b>"first(1) { if (((!is_list(1))&amp;&amp;(!is_array(1)))) return 1; start:=0; f:=nil; for (x in 1) if ((start==0)) { start:=1; f:=x; break; }; ; return f; }"</b>
<b>bodyof 1</b>	<i>raises an error</i>

#### **structof operator**

The **structof** is used to get the meta-type of a **struct** atom. The meta-type of a **struct** atom is the list of the attributes of this **struct**. For instance, the meta-type of **struct(a : 1, b : "hello")** is the list composed of the two attribute **a** and **b**. The following expression **structof struct(a : 1, b : "hello")** returns **list("a", "b")**.

<i>General Information</i>	
Operator	<b>structof</b>
Syntax	<b>structof</b> <i>expr</i>
Type	unary
Operand Type	<b>struct</b>
Result Type	a list of <b>strings</b>
Function	returns the meta-type of the operand

<i>Expression Examples</i>	
<b>structof struct(alpha : 1, beta : 2)</b>	<b>list("alpha", "beta")</b>
<b>structof first(select struct(x: x.firstname) from Person x)</b>	<b>list("x")</b>
<b>structof 1</b>	<i>raises an error</i>

#### **[!] operator**

The **[!]** is used to get the length (or size) of an ordered or unordered collection, a string or a struct. For instance, **"hello"[!]** returns 5, while **list(1, 2, 3)[!]** returns 3. Note that this operator is far more efficient than the **strlen**

library function.

<i>General Information</i>	
Operator	[!]
Syntax	<i>expr</i> [!]
Type	unary
Operand Type	<b>string</b> , <b>collection</b> or <b>struct</b>
Result Type	a <b>int</b>
Function	returns length of the operand

<i>Expression Examples</i>	
(select Person) [!]	the number of person instances
(struct(a: 1, b:2, c: "hello")) [!]	3
("hello"+"world") [!]	10
"hello"+"world" [!]	raises an error

### import operator

The **import** operator is used to import an OQL file in the current OQL session. Its operand is a string which denote the absolute path (i.e. beginning with a /) or relative path (i.e. not beginning with a /) of the file to import. When the path is relative, the OQL interpreter will look in every directories pointed by the EYEDB configuration variable **oqlpath**. By default, **oqlpath** is equal to **%root%/etc/oql**. If the file name has no **.oql** extension, the OQL interpreter will automatically adds one.

<i>General Information</i>	
Operator	<b>import</b>
Syntax	<b>import</b> <i>expr</i>
Type	unary
Operand Type	<b>string</b>
Result Type	a <b>string</b>
Function	import the file

<i>Expression Examples</i>	
<b>import</b> "stdlib"	"/usr/local/eyedb/etc/so/stdlib.oql"
<b>import</b> "roudoudou"	raises an error: cannot find file 'roudoudou'

## 5.31 Selection Statements

The selection statement is based on the **if/else** constructs.

**Syntax:** **if** ( *cond\_expr* ) *statement1* [**else** *statement2*]

where *cond\_expr* is a boolean expression, and *statement1* and *statement2* may be any statement: an expression statement, an iteration statement, a compound statement, a selection statement, a function definition statement, a jump statement or an empty statement.

**Semantics:** if the boolean expression *cond\_expr* is evaluated to **true**, the statement *statement1* is executed. Otherwise, if an **else** part is there, the statement *statement2* is executed. The statements 1 and 2 may be any statement: an expression statement, iteration statement, compound statement, selection statement, function definition statement, jump statement or empty statement.

Note that a selection statement does not return any atom

The following table presents several examples of **if/else** statements:

<i>if/else Statement Examples</i>
<code>if (true) a := 1;</code> the variable <code>a</code> is assigned to 1
<code>if (1) b := 2;</code> an error is raised: <code>boolean expected for condition</code>
<code>if (check(10) &gt; 54) {a := 1; b := 2;} else {c := 2; d := 2}</code> here compound statements are used, because several expression statements need to be executed
<code>if ((check(10) &gt; 54    alpha &lt; 2) &amp;&amp; beta &gt; 2.3 ) {callme(2);}</code> use of a composite conditional expression
<code>if (a == 1) b := 2; else if (b == 3) {c := 2; return 4;} else if (check(1)) return 2; else return 3;</code> selection statements are combined

## 5.32 Iteration Statements

The iteration statements are based on the following constructs:

- `while`
- `do/while`
- `for` C, C++, Java form
- `for` collection form

### `while` statement

**Syntax:** `while ( cond_expr ) statement`

where *cond\_expr* is a boolean expression, and *statement* any statement: an expression statement, an iteration statement, a compound statement, a selection statement, a function definition statement, a jump statement or an empty statement.

**Semantics:** The statement is executed while the boolean expression *cond\_expr* is evaluated to `true`. Note that a `while` statement does not return any atom

The following table presents several examples of `while` statements:

<i>while Statement Examples</i>
<code>while (true) a++;</code> <code>a</code> is increment definitively!
<code>while (n--) a++;</code> an error is raised: <code>boolean expected, got integer</code>
<code>while (n-- &gt; 0) a++;</code> this is better
<code>while (n++ &lt;= 100    stop) {if (!perform(a++)) break; check(a);}</code> note the usage of a compound statement and of the <code>break</code>
<code>while (name != "john") {l := (select Person.name = name); name := get_name();}</code> <i>no comments</i>

### `do/while` statement

**Syntax:** `do statement while ( cond_expr )`

where *cond\_expr* is a boolean expression, and *statement* any statement: an expression statement, an iteration statement, a compound statement, a selection statement, a function definition statement, a jump statement or an empty statement.

**Semantics:** The statement is executed at least once. Then while the boolean expression *cond\_expr* is evaluated to `true`, the statement is executed. Note that a `do/while` statement does not return any atom

The following table presents several examples of **do/while** statements:

<i>do/while Statement Examples</i>
do a++; while (true); a is increment definitively!
do a=+; while (n--); an error is raised: <code>boolean expected, got integer</code>
do a++; while (n-- > 0); this is better
do {if (!perform(a++)) break; check(a);} while (n++ <= 100    stop); note the usage of a compound statement and of the <code>break</code>
do {l := (select Person.name = name); name := get_name();} while (name != "john"); <i>no comments</i>

### C-for statement

**Syntax:** `for ( [expr1] ; [cond_expr] ; [expr2] ) statement`

where *cond\_expr* is a boolean expression, *expr1* and *expr2* are any expressions and *statement* any statement: an expression statement, an iteration statement, a compound statement, a selection statement, a function definition statement, a jump statement or an empty statement.

**Semantics:** The expression *expr1* is evaluated. While the boolean expression *cond\_expr* is evaluated to `true`, the statement is executed and the expression *expr2* is evaluated.

Note that a **C-for** statement does not return any atom

The following table presents several examples of **C-for** statements:

<i>C-for Statement Examples</i>
for (x := 0; x < 100; x++) a++; increment a an hundred times
for (x := 0, y := 1; x < 100 && check(y); x++) {y := get(y); if (y == 9999) break;} a more complex example
for (x := 100; x; x--) perform(x); raises an error: <code>boolean expected, got integer</code>
for (x := 0;;) doit(); note that there is neither a conditionnal expression, nor a second expression
for (;;) doit(); same as <code>while(true)</code>

### collection-for statement

**Syntax:** `for ( var in expr ) statement`

where *cond\_expr* is a boolean expression, *var* denotes the name of a variable, *expr* is an expression of type collection and *statement* any statement: an expression statement, an iteration statement, a compound statement, a selection statement, a function definition statement, a jump statement or an empty statement.

**Semantics:** For each element in the collection denoted by *expr*, the variable *var* is assigned to this element and the statement is executed. Note that a **collection-for** statement does not return any atom

The following table presents several examples of **collection-for** statements:

<i>collection-for Statement Examples</i>
--

for (x in list(1, 2, 3)) a += x; increments a with 1, 2 and 3
for (x in (select Person)) names += x.name; concatenates all the person names
for (x in (select Person.name = "john")) if (x.age < 10    x.spouse.age < 10) throw "cannot mary children!!"; a moralistic example
for (x in 1) doit(); raises an error: boolean expected, got integer

### 5.33 Jump Statements

There are two jump statements based on the keywords **break** and **return**.

#### break Statement

The syntax for **break** statement is:

**break** *statement*

where *statement* is an optional integer expression statement indicating the break level. If not specified, the break level is

1. The **break** statement may appear only within an iteration statement: **while**, **do/while**, **for** statements.

break Statement Examples	
Statement	Comments
break;	raises an error: break operator <<break; >> : level 1 is too deep.
break 3;	raises an error: break operator <<break; >> : level 3 is too deep.
for (x := 0; ; x++) if (x == 30) break;	break the loop when x is equal to 30
while (true) {x++; if (!check(x)) break;}	break the loop when check(x) is not true
while (true) {x++; if (!check(x)) break 2;}	raises an error: break operator <<break; >> : level 2 is too deep.
while (true) {x++; while (x < 100) if (!check(&x)) break 2;}	break the two loop when check(&x) is not true

#### return Statement

The syntax for **return** statement is: **return** [*statement*]

where *statement* is an optional expression statement indicating the atom to return. The atom to return may be of any type. If not specified, no atom is returned. The **return** statement may appear only within a function definition statement.

return Statement Examples	
Statement	Comments
return;	raises an error: return operator <<return; >> : return must be performed in a function
function fact(n) {return n * fact(n-1);}	ok.
function f(x) {if (g(x)) return x+1; return x+2;}	ok.
function f(b) {if (b) return list(1, 2, 3); return bag(1);}	ok.
function f(b) {if (b) return list(1, 2, 3); return bag(1);}	ok.
define as f(x) (if (x) return x)	raises an error: syntax error

### 5.34 Function Definition Statements

As introduced previously, OQL supports functions. There are two sorts of functions definition syntax: function definition expression and function definition statements. The first ones, exposed in Section 5.26 can contain only one expressions. That means that they cannot include neither selection, neither statements, neither jump statements. Furthermore, as only one expression is allowed, functions that need several expressions, one must use the comma sequencing operator to separate the expressions, thus making this code not always readable.

We introduce here the more general form of function definitions which overrides the limitations of the previous form. The general form of function definition statements is: **function** *identifier* ([*arglist*]) *compound\_statement*

1. *identifier* denotes any valid OQL identifier, except a keyword
2. *arglist* is an optional comma-separated list of identifiers optionally followed, for default arguments, by a “?” and an *expr*, for instance:  
(*var1*, *var2*, *var3*? *expr*, *var4*? *expr*)
3. *compound\_statement* is a optional semicolon-separated list of statements surrounded by braces.

For instance:

```
function f(x, y, z ? oql$maxint) {
  if (x > y)
    throw "error #1";
  return x - y * 2 / z;
}
```

#### Argument Types/Return Type

Functions are not typed. That means that neither the return type nor the argument types may be given. It is why there is no function overloading mechanisms. To take benefit of the overloading mechanisms, one must use methods.

Nevertheless, it is possible to add type checking by using library functions such as `is_int`, `is_string`... combined with the `assert` or the `assert_msg` library functions. For instance, to check that the first argument is an integer and the second one a collection:

```
function doit(n, coll) {
  assert_msg(is_int(n), "doit: argument #1: integer expected");
  assert_msg(is_coll(coll), "doit: argument #2: collection expected");
  // body of the function
}
```

The `assert_msg` check that its first argument is equal to `true`, otherwiser an exception containing the second argument string is thrown:

```
doit(1, list(1, 2, 3)); // ok

doit(1.2, list(1)); // raises the error:
// assertion failed: 'doit: argument #1: integer expected'
```

#### Arguments in, out and inout

Furthermore, one cannot specify that an argument is an input argument (`in`), an output argument (`out`) or an input/output argument (`inout`). In a function call, expressions and variables are always passed by value not by reference, this means that the call to “`perform(x, y)`” cannot modify neither `x` nor `y`. (*In fact, yes it can! It is explained below. But forget it for now*).

So, to modify variable through a function call, one needs to give the reference (or address) of this variable, not its value. In this case, the function must execute specific code dealing with address variables instead of their values.

The `refof` operator, introduced in a previous section, gives the reference of an identifier. Remember that the expression `refof x` returns the identifier `x`. To make a function call using references one must do: `swap(refof x, refof y)` or the equivalent more compact form `swap(&x, &y)`.

Contrary to C++, reference manipulation is not transparent in OQL: to access the value of a reference, one must use the `valof` operator (i.e. `*` operator). The `swap` function which swaps its two `inout` arguments has already been introduced:

```
function swap(rx, ry) {
  v := *rx;
  *rx := *ry;
  *ry := v;
}
```



The arguments have been prefixed by `r` to indicate that they are references. So, the function call `swap(&x, &y)` will swap properly the variables `x` and `y`.

One can add type checking in the `swap` function, as follows:

```
function swap(rx, ry) {
  assert_msg(is_ident(rx), "swap: argument #1 identifier expected");
  assert_msg(is_ident(ry), "swap: argument #2 identifier expected");
  v := *rx;
  *rx := *ry;
  *ry := v;
}
```

### Return Value

By default, a statement-oriented function returns no atom. To make a function returning an atom, one must use the `return` statement previously introduced. As a function has no specified returned type, it may contain several `return` statements returning atom of different types:

```
function perform(x) {
  if (x == 1)
    return "hello";
  if (x == 2)
    return list(1, 2, 3) + list(4, 20);
  if (x == 3)
    return 2;
  if (x == 4)
    return 'a';
}

alpha := perform(1); // alpha is equal to "hello"
alpha := perform(3); // alpha is equal to 2
alpha := perform(8); // alpha is equal to nil
```

### Default Arguments

OQL provides support for default arguments in a function definition statement. The syntax for a default argument is: “`var ? expr`” or “`var := expr`”.

As in C and C++, the arguments with a default value must not be followed by any argument with default values. For instance, `function f(x, y, z := "alpha")` is valid while `function f(x, y, z := "alpha", t)` is not valid.

### Unval Arguments

Sometimes, it is interesting to prevent the evaluation of some input arguments. For instance, let the function `if_then_else` which takes three arguments:

1. `cond`: a boolean expression,
2. `then_expr`: expression of any type; is evaluated and returned if and only if the condition is evaluated to `true`
3. `else_expr`: expression of any type; is evaluated and returned if and only if the condition is evaluated to `false`

It is clear that the following function definition:

```
function if_then_else(cond, then_expr, else_expr) {
  if (cond)
    return then_expr;
  return else_expr;
}
```

is not correct as, although it returns the correct expression, the `then_expr` and the `else_expr` will be evaluated. For instance, `if_then_else(x < 10, ::a := 2, ::b := 3)` will return 2 if `x` is less than 10, otherwise it will return 3, but in any case, `a` will be assigned to 2 and `b` will be assigned to 3.

So, one needs a way to tell the interpreter that we do not want to evaluate the second and the third argument. The special character `|` before an argument means that this argument must not be evaluated. In this case, this argument is substituted by the string representation of the expression. For instance, let the function `if_then_else`:

```
function if_then_else(cond, |then_expr, |else_expr) {
  // ...
}
```

when performing the call “`if_then_else(x < 10, ::a := 2, ::b := 3)`”:

1. the value of `cond` in the body of the function will be `true` or `false`,
2. the value of `then_expr` in the body of the function will be “`::a:=2`”
3. the value of `else_expr` in the body of the function will be “`::b:=3`”

The correct implementation of this function is as follows:

```
function if_then_else(cond, |then_expr, |else_expr) {
  if (cond)
    return eval then_expr;
  return eval else_expr;
}
```

### Scope of Variables

In the body of a function definition, every variable on the left side of an assignment has a local scope except if this variable is prefixed by the global scope operator `::`. That means, that after the following statement sequence:

```
a := 2;

function doit() {
  a := 1;
}
```

the variable `a` is still equal to 2. While after:

### Recursivity

```
a := 2;

function doit() {
  ::a := 1;
}
```

the variable `a` is equal to 1.

### Particularity

One can define a statement-oriented function inside the body of a statement-oriented function, for instance:

```
function compute(amount_in_euro, usdollar_per_euro) {

  function euro2usdollar(euro, usd ? usdollar_per_euro) {
    return euro * usd;
  }

  x := euro2usdollar(euro * 1.24);
  x += euro2usdollar(1000);

  return x * .120;
}
```

Note that the function defined in the body of the function `compute` has a global scope, that means that after one execution of `compute` the function is available at the global level of the OQL session. It is possible, that in a future version, the functions defined in the body of a function definition will have a local scope.

### The `oql$functions` Variable

The `oql$functions` value is a list whose elements are the name of all the OQL functions defined in the current OQL session. Each you add a user function, this variable is updated. At the beginning of a session, the value of `textttoql$functions` is:

```
list(is_int, is_char, is_double, is_string, is_oid, is_num, is_bool, is_bag,
     is_set, is_array, is_list, is_coll, is_struct, is_empty, void, assert,
     assert_msg, min, max, first, last, cdr, count, interval, sum, avg, is_in,
     distinct, flatten, flatten1, tolower, toupper, tocap, toset, tolist,
     tobag, toarray, listtoset, bagtoset, arraytoset, listtobag, settobag,
     arraytobag, bagtolist, settolist, arraytolist, bagtoarray, settoarray,
     listtoarray, strlen, substring, forone, forall, delete_from, get_from,
     ifempty, null_ifempty, getone)
```

For instance, to put the complete definition of all these functions into the variable `functionString`:

```
functionString := "";
for (x in oql$functions)
  functionString += "FUNCTION " x + ": " + bodyof x + "\n";
```

The next section provides a few statement-oriented and expression-oriented function definitions.

## 6 Quick Reference Manual

This OQL quick reference manual presents concise information about the builtin and library functions and methods, the special variables, and the `eyedboql` tool. The standard library source code is presented and it provides a quick reference card containing all the language constructs.

### 6.1 Builtin and Library Functions and Methods

OQL provides a few builtin and library functions. The builtin functions are written in C++ and cannot be overridden while the library functions are written in OQL and may be redefined by the user. The code for the library functions can be found in the section *The Standard Library Package*. The EYEDB system classes `object`, `database`, `connection` and `EyeDB` contains builtin class and instance methods that can be accessed from OQL. Some of these methods are briefly introduced in this section.

#### Type Predicat Functions

```

is_int(x)      : returns true if x is an int
is_double(x)   : returns true if x is a double; otherwise, returns false
is_string(x)   : returns true if x is a string; otherwise, returns false
is_oid(x)      : returns true if x is an oid; otherwise, returns false
is_num(x)      : returns true if x is an number (int, float or char); otherwise, returns false
is_bool(x)     : returns true if x is a bool; otherwise, returns false
is_bag(x)      : returns true if x is a bag; otherwise, returns false
is_set(x)      : returns true if x is a set; otherwise, returns false
is_array(x)    : returns true if x is a array; otherwise, returns false
is_list(x)     : returns true if x is a list; otherwise, returns false
is_coll(x)     : returns true if x is a collection; otherwise, returns false
is_struct(x)   : returns true if x is a struct; otherwise, returns false
is_empty(x)    : returns true if x is nil; otherwise, returns false

```

#### Collection Conversion Functions

The collection conversion functions take one collection argument and convert this collection to another collection type and returns the converted collection.

```

toset(coll)           : converts coll to a set
tolist(coll)          : converts coll to a list
tobag(coll)           : converts coll to a bag
toarray(coll)         : converts coll to a array
listtoset(coll)       : checks that coll is a list then converts coll to a set
bagtoset(coll)        : checks that coll is a bag then converts coll to a set
arraytoset(coll)      : checks that coll is a array then converts coll to a set
listtobag(coll)       : checks that coll is a list then converts coll to a bag
settobag(coll)        : checks that coll is a set then converts coll to a bag
arraytobag(coll)      : checks that coll is a array then converts coll to a bag
bagtolist(coll)       : checks that coll is a bag then converts coll to a list
settolist(coll)       : checks that coll is a set then converts coll to a list
arraytolist(coll)     : checks that coll is a array then converts coll to a list
bagtoarray(coll)      : checks that coll is a bag then converts coll to a array
settoarray(coll)      : checks that coll is a set then converts coll to a array
listtoarray(coll)     : checks that coll is a set then converts coll to a array

```

#### Sort Functions

These functions are used to sort collection of sortable atom of homogeneous types: `int`, `char`, `float` or `string`.

```

sort(coll)           : coll must a be a collection of homogeneous sortable atoms;
                      : sorts and returns this collection
rsort(coll)          : coll must a be a collection of homogeneous sortable atoms;
                      : reverse sorts and returns this collection
isort(coll, idx)     : coll must a be a collection of list or array
                      : of homogeneous sortable atoms;
                      : idx must be of int type;

```

sorts the collection of collections according to the  
`#idx` element of the inner collection  
`risort(coll, idx)` : same as previous function, but perform a reverse sort

### Collection Miscellaneous Functions

`first(coll)` : returns the first element of `coll`  
`car(coll)` : returns the first element of `coll`  
`last(coll)` : returns the last element of `coll`  
`cdr(coll)` : returns all elements of `coll` but the first  
`getn(coll, n)` : returns at most `n` elements of `coll`  
`count(coll)` : returns the count of elements of `coll`  
 identical to `coll[!]`, but less efficient  
`sum(coll)` : returns the sum of the numbers of `coll`  
`avg(coll)` : returns the float average of the numbers `coll`  
`distinct(coll)` : eliminates duplicates of `coll`  
`flatten(coll)` : recursive flattening of `coll`  
`flatten1(coll)` : one level flattening of `coll`  
`min(coll)` : returns the minimal number of `coll`  
`max(coll)` : returns the maximal number of `coll`  
`forone(coll, f, data)` : if `f(e, data)` for one element `e`  
 of `coll`, returns `true`;  
 otherwise returns `false`;  
`forall(coll, f, data)` : if `f(e, data)` for all element `e`  
 of `coll`, returns `true`;  
 otherwiser returns `false`;

### String Function Functions

`tolower(str)` : converts (and returns) string `str` into lowercase  
`toupper(str)` : converts (and returns) string `str` into uppercase  
`tocap(str)` : converts the first character and each character following  
 a \_ of `str` into an uppercase  
`strlen(str)` : returns the length of `str`;  
 same as `str[!]`, but less efficient  
`substring(str, from, len)` : returns the sub-string of `str`  
 from the `#from` to the  
`from+len` characters;  
 same as `str[from:from+len]` but less efficient

### Query Functions

`delete_from(class)` : deletes all the instances of a given class  
`get_from(class)` : gets all the instances of a given class

### Useful Functions

`assert(cond)` : throws an exception is `cond` is not `true`  
`assert_msg(cond, msg)` : throws the exception message `msg` if `cond` is not `true`  
`interval(from, to)` : returns a list composed of the number from `from` to `to`

### Native Methods of the Class object

The native methods of the class `object` allows us to perform a few action such as getting the oid of an instance `getOid()`, getting the database of an instance `getDatabase()` or converts the instance to its string representation `toString()`. For instance, to apply this last method to the first `Person` instance: `first(select Person).toString()`.

All the native methods of the class `object` are instance methods.

`oid getOid()` : returns the `oid` of the object  
`string toString()` : returns the string representation of the object  
`database *getDatabase()` : returns the `database` instance of the object  
`void setDatabase(in database *)` : changes the `database` of the object

```

void store()                : stores the object in the database
object *clone()             : clones the object; returns the clone
int getCTime()              : returns the creation time of the object (seconds from 1/1/1970)
int getMTime()              : returns the last modification time of the object
string getStringCTime()     : returns the string representation of the creation time of the object
string getStringMTime()     : returns the string representation of the creation time of the object
bool isRemoved()            : returns true if the object is removed; false otherwise
bool isModify()             : returns true if the object is modified; false otherwise

```

### Native Methods of the Class connection

All the native methods of the class **connection** are instance methods. an object obtained using the **new** operator, They can be applied on a **database** object that can be either the current connection `oql$db->getConnection()` either an object obtained using the **new** operator, for instance: `new <> connection()`.

```

void open()                : opens a new connection with default host and port
void open(in string host, in string port) : opens a new connection using host and port
void close()               : closes the connection

```

### Native Methods of the Class database

The following methods are the instance methods of the class **database**: They can be applied on a **database** object that can be either `oql$db` either an object obtained using the **new** operator, for instance: `new <> database(name : "foo")`.

```

void open(in connection *conn,
          in int mode)        : opens a new database using the connection
                              conn and the open flag mode mode
void open(in connection *conn,
          in int mode,
          in string userauth,
          in string passwdauth) : opens a new database using the connection conn,
                              the open flag mode mode and the authentication
                              userauth/passwdauth
void close()                 : closes the database
connection *getConnection()  : returns the connection tied to the database
int getOpenMode()            : returns the open flag mode of the database
int getVersionNumber()       : returns the version number of the database
string getVersion()          : returns the string version of the database

void removeObject(in oid)    : removes the object whose oid is given

void transactionBegin()      : begins a new transaction
void transactionBegin(in string mode) : begins a new transaction in mode mode
void transactionCommit()     : commits the current transaction
void transactionAbort()      : abort the current transaction

bool isInTransaction()       : returns true if a transaction is in progress;
                              false otherwiser

```

### Native Methods of the Class EyeDB

All the native methods of the class **EyeDB** are class methods.

```

string getConfigValue(in string s) : returns the string value of the configuration variable s
int getVersionNumber()             : returns the EYEDB current version number
string getVersion()                : returns the EYEDB current stringversion
string getArchitecture()           : returns the architecture of the current server
string getDefaultCompiler()        : returns the C++ compiler used to compile the current server

```

## 6.2 Special Variables

The following variables are predefined or have a special meaning:

```
oql$variables : list containing the name of all variables
oql$functions : list containing the name of all functions
oql$result    : the result atom of the last statement
oql$db        : object atom instance of the class database denoting the current database
oql$minint    : the minimal integer 0x8000000000000000
oql$maxint    : the maximal integer 0x7FFFFFFFFFFFFFFF
oql$minfloat  : the minimal float 4.94065645841246544e-324
oql$maxfloat  : the maximal float 1.79769313486231570e+308
```

## 6.3 The eyedboql Tool

**eyedboql** is a tool that allows you to execute interactively OQL statements. This tool is similar to the Oracle **sqlplus** and Sybase **isql** well known tools.

### Running eyedboql

The simplest way to run **eyedboql** is to type **eyedboql** as follows (assuming that **\$** is your shell prompt):

```
$ eyedboql
Welcome to eyedboql.
  Type '\help' to display the command list.
  Type '\copyright' to display the copyright.
?
```

The string “? ” is the default prompt for **eyedboql**.

In an **eyedboql** session, anything you type is interpreted as an OQL construct (or a part of an OQL construct), Nevertheless, if the first word of an input line begins with the escape character “ ”, this word is interpreted as an **eyedboql** internal command. There are about 30 internal commands, but you need to know only a few of them to use **eyedboql**.

The purpose of the main internal commands is:

- to create or delete databases,
- to open a database,
- to begin, commit or abort a transaction,
- to set the current user and password,
- to execute the contents of a file,
- to display the string representation of an object,
- to display the HTML representation of an object in a WEB browser,
- to change the prompts and the escape character.

### Executing Statements

**eyedboql** allows us to execute OQL statements in an interactive way. For instance, to perform the addition **1+3**:

```
$ eyedboql
Welcome to eyedboql.
  Type '\help' to display the command list.
  Type '\copyright' to display the copyright.
? 1+3;
= 4
?
```

The string “= ” precedes the result atom (if any) of your statement; in the current example, the result atom is the evaluation of the expression statement **1+3**;

### Execution Process

Each complete statement typed is sent to the OQL interpreter, A complete statement has a special meaning: it is any sequence of characters:

- a. that end with a semi-colon and

- b. which parenthesis are balanced and
- c. which brackets are balanced and
- d. which braces are balanced.

While the statement is not complete, **eyedboql** prompts the “second prompt” (“>> ” by default) after each newline. Once the statement is complete, it is sent to the OQL interpreter, then the atom result is display (if any) after the string “= ” and the main prompt “? ” is displayed again.

For instance, the input sequence of characters “1+newline3newline;” gives:

```
? 1+
>> 3
>> ;
= 4
```

while the input sequence “{ a := 1+3;newlinec := 2+94;newlined := a+c}” gives:

```
? { a := 1+3;
>> c := 2+94;
>> d := a+c}
?
```

Note that no “= *result atom*” is echoed because a compound statement does not return any atom.

This last example:

```
? while (true) {
>>   a++;
>>   b++;
>> }
>>
>> ;
?
```

shows the necessity of typing a semicolon after the **while** statement although a **while** statement does not need to end by a semi-colon in the OQL specifications.

## Getting Started

By default in an **eyedboql** session, the database **EYEDBDBM** is opened in read-only mode. To use another database, one must use either the command line option **-db** either the **open** internal command.

To start to play with **eyedboql** one needs to know the following five internal commands:

1. **open database [rw|ro] local trsless**: opens the *database* in read-write (**rw**) or read-only (**ro** or no option) mode. If **local** is there, *database* is opened in a local mode. If **trsless** is there *database* is opened in transaction-less mode. For instance “**open foo rw**” opens the database **foo** in read-write mode.
2. **print [sequence of oids] other options**: if a sequence of oids is given: loads the object corresponding to each **oid** and displays its string representation, if no sequence of oids is given: loads the object corresponding to each **oid** returned by the last statement and displays its string representation. For instance, “**select Person;**” will return a **bag** containing the **oid** of each **Person** instance. The internal command “**print**” typed after that will load and display of the corresponding **Person** instances. The *other options* are one or more space-separated of the followings:

```
full      : loads and displays object using the full recursive mode
ctime     : displays the creation time of the object
mtime     : displays the last modification time of the object
contents  : displays the contents of collections
native    : displays the native attributes
all       : means “ctime mtime contents native”
```



For instance “

`print full contents`” will load and display the objects and their collection contents in a full recursive mode.

3.  
`commit`: commits the current transaction
4.  
`abort`: aborts the current transaction
5.  
`quit` or `^D`: quits the current `eyedboql` session

Note that a transaction is started automatically before the first complete statement of the session or before the complete statement immediately following

`commit` or

`abort` internal command.

Here is a commented example showing the use of these internal commands:

`run eyedboql`:

`$ eyedboql`

Welcome to eyedboql.

Type ‘\help’ to display the command list.

Type ‘\copyright’ to display the copyright.

*open the database person in read-write mode:*

`? \open person rw`

*get the first person whose name is "john" and display it:*

`? john := first(select Person.name = "john");`

`= 66373.12.4008447:oid`

`? \print`

```
66373.12.4008447:oid Person = {
  name = "john";
  age = 32;
  addr Address = {
    street = "clichy";
    town = "Paris";
    country = NULL;
  };
  cstate = Sir;
  *spouse 66891.12.2738687:oid;
  cars set<Car*> = set {
    name      = "";
    count     = 4;
    dimension = 1;
    reference = true;
    magorder  = 4;
  };
  children array<Person*> = array {
    name      = "";
    count     = 0;
    range     = [0,0[;
    dimension = 1;
    reference = true;
    magorder  = 4;
  };
  x = NULL;
};
```

*change the name of john to "JOHNNY":*

`? john.name := "JOHNNY";`

`= "JOHNNY"`

retrieve the person whose name is "JOHNNY" and compares it to john using **assert** : all is fine, no error is raised!

```
? assert(john = first(select Person.name = "JOHNNY"));
```

abort the transaction and look for the person whose name is "JOHNNY": no person is returned! this is ok as the transaction was aborted:

```
? \abort
? select Person.name = "JOHNNY";
= list()
```

change the name of john to "JOHNNY" again and commit the transaction:

```
? john.name := "JOHNNY";
= "JOHNNY"
? \commit
```

then retrieve again the person whose name is "JOHNNY" and compare it to john using **assert**: all is fine, no error is raised!

```
? assert(john = first(select Person.name = "JOHNNY"));
```

quit **eyedboql** session

```
? \quit
$
```

We are going to conclude this section by this important note:

as introduced previously, the current transaction will be committed (resp. aborted) by a **commit** (resp. **abort**) command.

If you quit **eyedboql** before committing (resp. aborting) the transaction, it will be automatically aborted, so all your changes will be lost.

This is the default behaviour. This behaviour can be changed by using the **commitonclose** internal command.

## Command Line Options

The **eyedboql** command line options are as follows:

### Program Options:

-d <name>, -database=<name>	Database name
-r, -read	Open database in read mode
-w, -read-write	Open database in read/write mode
-s, -strict-read	Open database in strict read mode
-l, -local	Open database in local mode
-c <command>, -command=<command>	OQL command to execute
-p, -print	Display all the objects loaded
-full	Full recursive mode is used to display objects
-commit	Commits the current transaction on close
-i, -interact	Enter interpreter after executing file or commands
-e, -echo	Echo each command
-admin	Open database in admin mode
-h, -help	Display this message
<file>	File(s) to execute

### Common Options:

-U <user> @, -user=<user> @	User name
-P [<passwd>], -passwd[=<passwd>]	Password
-host=<host>	eyedbd host
-port=<port>	eyedbd port
-inet	Use the tcp_port variable if port is not set
-dbm=<dbmfile>	EYEDBDBM database file
-conf=<conffile>	Configuration file
-logdev=<logfile>	Output log file
-logmask=<mask>	Output log mask
-logdate=on off	Control date display in output log

<code>-logtimer=on off</code>	Control timer display in output log
<code>-logpid=on off</code>	Control pid display in output log
<code>-logprog=on off</code>	Control progname display in output log
<code>-error-policy=&lt;value&gt;</code>	Control error policy: <code>status exception abort stop echo</code>
<code>-trans-def-mag=&lt;magorder&gt;</code>	Default transaction magnitude order
<code>-arch</code>	Display the client architecture
<code>-v, -version</code>	Display the version
<code>-help-eyedb-options</code>	Display this message

For instance, to execute the statement “`delete_from(Person)`” on the database `person`:

```
$ eyedboql -d person -w -c "delete_from(Person)"
$
```

To execute the command “`persons := (select Person)`” and then enter the interactive mode of `eyedboql`:

```
$ eyedboql -d person -w -c "persons := (select Person)" -i
Welcome to eyedboql.
Type '\help' to display the command list.
Type '\copyright' to display the copyright.
?
```

To execute the file `mylib.oql`:

```
$ eyedboql -d person -w mylib.oql
```

## 6.4 The Standard Library Package

The `stdlib.oql` file contains a few basic library functions. It can be found in the directory `libdir/eyedb/oql`. It is automatically imported when opening an OQL session. The following code, extracted from this file, provides an interesting way to understand OQL.

```
//
// minimal and maximal integer values
//

oql$maxint := 0x7FFFFFFFFFFFFFFF;
oql$minint := 0x8000000000000000;

nulloid := 0:0:0:oid;
NULLOID := 0:0:0:oid;

//
// type predicat functions
//

define is_int(x)    as (typeof x == "integer");
define is_char(x)   as (typeof x == "char");
define is_float(x)  as (typeof x == "float");
define is_string(x) as (typeof x == "string");
define is_oid(x)    as (typeof x == "oid");
define is_object(x) as (typeof x == "object");
define is_num(x)    as (is_int(x) || is_float(x) || is_char(x));
define is_bool(x)   as (typeof x == "bool");
define is_bag(x)    as (typeof x == "bag");
define is_set(x)    as (typeof x == "set");
define is_array(x)  as (typeof x == "array");
define is_list(x)   as (typeof x == "list");
define is_coll(x)   as (is_list(x) || is_array(x) || is_set(x) || is_bag(x));
define is_struct(x) as (typeof x == "struct");
define is_empty(x)  as (typeof x == "nil");

//
// void(x): evaluates argument and returns nil
//
```

```

define void(x) as (x, nil);

function assert(|cond) {
  r := eval cond;
  if (!r)
    throw "assertion failed: '" + cond + "'";
}

function assert_msg(|cond, msg) {
  r := eval cond;
  if (!r)
    throw "assertion failed: '" + msg + "'";
}

//
// min(l): returns the minimal integer in a collection
//

function min(l) {
  msg := "function min(" + (string l) + "): ";
  if (!is_coll(l))
    throw msg + "collection expected";

  m := oql$maxint;

  for (x in l) {
    if (x != null) {
      if (!is_num(x))
        throw (msg + "numeric expected");
      if (x < m)
        m := x;
    }
  }

  return m;
}

//
// max(l): returns the maximal integer in a collection
//

function max(l) {
  msg := "function max(" + (string l) + "): ";
  if (!is_coll(l))
    throw msg + "collection expected";
  m := oql$minint;
  for (x in l) {
    if (x != null) {
      if (!is_num(x))
        throw (msg + "numeric expected");
      if (x > m)
        m := x;
    }
  }
  return m;
}

//
// first(l): returns the first element in a list or array
//

function first(l) {

```

```

if (!is_coll(l)) // if (!is_list(l) && !is_array(l))
    throw "function first: collection expected";

start := 0;
f := nil;
for (x in l)
    if (start == 0) {
        start := 1;
        f := x;
        break;
    }
return f;
}

car := &first;

//
// last(l): returns the last element in a list or array
//

function last(l) {
    if (!is_coll(l)) // if (!is_list(l) && !is_array(l))
        throw "function last: list or array expected";

    f := nil;
    for (x in l)
        f := x;
    return f;
}

//
// cdr(l): returns all elements in a collection except the first one
//

function cdr(l) {
    if (!is_coll(l)) // if (!is_list(l) && !is_array(l))
        throw "function cdr: list or array expected";

    r := list();
    n := 0;
    for (x in l) {
        if (n != 0)
            r += x;
        n++;
    }
    return r;
}

//
// getn(l, n): returns all elements in a collection
//

function getn(l, n) {
    rl := list();
    cnt := 0;

    for (x in l) {
        if (cnt++ >= n)
            break;
        rl += x;
    }

    return rl;
}

```

```

}

//
// getrange(l, f, nb): returns all elements in a collection from element from
//
// identical to l[f:f+nb]
//

function getrange(l, f, nb) {
  if (!is_list(l) && !is_array(l))
    throw "function getrange: list or array expected";

  rl := list();
  cnt := 0;

  max := f + nb;

  for (x in l) {
    if (cnt >= max)
      break;

    if (cnt >= f)
      rl += x;

    cnt++;
  }

  return rl;
}

//
// count(l): returns element count of a collection
//

function count(l) {
  if (typeof l == "nil")
    return 0;
  if (!is_coll(l))
    throw "function count: collection expected, got " + typeof(l);
  return l[!];
}

//
// interval(x, y): constructs an integer list bounded by 'x' and 'y'
//

function interval(x, y) {
  n := x-1;
  l := list();
  while (n++ < y)
    l += n;
  return l;
}

//
// sum(l): returns the sum of collection elements
//

function sum(l) {
  if (!is_coll(l))
    throw "function sum: collection expected";
  n := 0;
  for (x in l)

```

```

    n += x;
    return n;
}

//
// avg(l): returns the average of collection elements
//

function avg(l) {
    if (!is_coll(l))
        throw "function avg: collection expected";
    return float(sum(l))/count(l);
}

//
// is_in(l, z): returns true in element 'z' is in collection 'l'
//

function is_in(l, z) {
    for (x in l)
        if (x == z)
            return true;

    return false;
}

//
// distinct(l): returns distinct elements in a collection
//

function distinct(l) {
    if (is_list(l)) ll := list();
    else if (is_bag(l)) ll := bag();
    else if (is_array(l)) ll := array();
    else if (is_set(l)) ll := set();
    else throw "function distinct: collection expected";

    for (x in l)
        if (!is_in(ll, x))
            ll += x;

    return ll;
}

//
// flatten(l): full recursive flatten function
//

function flatten(l) {
    if (!is_coll(l))
        return l;
    ll := list();
    for (x in l)
        if (is_coll(x))
            ll += flatten(x);
        else
            ll += x;

    return ll;
}

//
// flatten1(l): 1-level recursive flatten function

```

```

//

function flatten1(l) {
  if (!is_coll(l))
    return l;

  ll := list();
  for (x in l)
    ll += x;
  return ll;
}

//
// tolower(s): returns lower case string
//

function tolower(s) {

  n := 0;
  x := "";
  delta := 'a' - 'A';

  while (s[n] != '\000') {
    if (s[n] >= 'A' && s[n] <= 'Z')
      x += string(char(s[n] + delta));
    else
      x += string(s[n]);
    n++;
  }

  return x;
}

//
// toupper(s): returns upper cased string
//

function toupper(s) {

  n := 0;
  x := "";
  delta := 'A' - 'a';

  while (s[n] != '\000') {
    if (s[n] >= 'a' && s[n] <= 'z')
      x += string(char(s[n] + delta));
    else x += string(s[n]);
    n++;
  }

  return x;
}

//
// tocap(s): returns capitalized word string
//

function tocap(s) {

  n := 1;
  x := "";
  delta := 'A' - 'a';

```



```

s := tolower(s);

if (s[0] >= 'a' && s[0] <= 'z')
  x += string(char(s[0] + delta));

while (s[n] != '\000') {
  if (s[n] == '_')
    x += string(char(s[++n] + delta));
  else
    x += string(s[n]);
  n++;
}

return x;
}

//
// Collection Conversion Functions
//

//
// General Conversion Functions
//

function toset(l) {
  if (!is_coll(l))
    throw ("function toset: collection expected, got " + typeof(l));

  if (!is_set(l)) {
    s := set();
    for (x in l)
      s += x;
    return s;
  }

  return l;
}

function tolist(l) {
  if (!is_coll(l))
    throw ("function tolist: collection expected, got " + typeof(l));

  if (!is_list(l)) {
    s := list();
    for (x in l)
      s += x;
    return s;
  }

  return l;
}

function tobag(l) {
  if (!is_coll(l))
    throw ("function tobag: collection expected, got " + typeof(l));

  if (!is_bag(l)) {
    s := bag();
    for (x in l)
      s += x;
    return s;
  }
}

```

```

    return l;
}

function toarray(l) {
  if (!is_coll(l))
    throw ("function toarray: collection expected, got " + typeof(l));

  if (!is_array(l)) {
    s := array();
    for (x in l)
      s += x;
    return s;
  }

  return l;
}

//
// toset family Conversion Functions
//

function listtoset(l) {
  if (!is_list(l))
    throw ("function listtoset: list expected, got " + typeof(l));
  return toset(l);
}

function bagtoset(l) {
  if (!is_bag(l))
    throw ("function bagtoset: bag expected, got " + typeof(l));
  return toset(l);
}

function arraytoset(l) {
  if (!is_array(l))
    throw ("function arraytoset: array expected, got " + typeof(l));
  return toset(l);
}

//
// tobag family Conversion Functions
//

function listtobag(l) {
  if (!is_list(l))
    throw ("function listtobag: list expected, got " + typeof(l));
  return tobag(l);
}

function settobag(l) {
  if (!is_set(l))
    throw ("function settobag: set expected, got " + typeof(l));
  return tobag(l);
}

function arraytobag(l) {
  if (!is_array(l))
    throw ("function arraytobag: array expected, got " + typeof(l));
  return tobag(l);
}

//

```

```

// tolist family Conversion Functions
//

function bagtolist(l) {
  if (!is_bag(l))
    throw ("function bagtolist: bag expected, got " + typeof(l));
  return tolist(l);
}

function settolist(l) {
  if (!is_set(l))
    throw ("function settolist: set expected, got " + typeof(l));
  return tolist(l);
}

function arraytolist(l) {
  if (!is_array(l))
    throw ("function arraytolist: array expected, got " + typeof(l));
  return tolist(l);
}

//
// toarray family Conversion Functions
//

function bagtoarray(l) {
  if (!is_bag(l))
    throw ("function bagtoarray: bag expected, got " + typeof(l));
  return toarray(l);
}

function setttoarray(l) {
  if (!is_set(l))
    throw ("function setttoarray: set expected, got " + typeof(l));
  return toarray(l);
}

function listtoarray(l) {
  if (!is_list(l))
    throw ("function listtoarray: list expected, got " + typeof(l));
  return toarray(l);
}

//
// strlen(s): same as s[!]
//

function strlen(s) {
  len := 0;
  while (s[len] != '\000')
    len++;
  return len;
}

//
// substring(str, f, len)
//

function substring(str, f, len) {
  s := "";
  n := 0;
  max := str[!] - f;
  while (n < len && n < max) {

```

```

        s += string(str[n+f]);
        n++;
    }

    return s;
}

//
// forone(l, fpred, data): returns true if and only if the function 'fpred'
// returns true for at least one element 'x' in list 'l'
//

function forone(l, fpred, data) {
    for (x in l)
        if (fpred(x, data)) return true;
    return false;
}

//
// forone(l, fpred, data): returns true if and only if the function 'fpred'
// returns true for all elements 'x' in list 'l'
//

function forall(l, fpred, data) {
    for (x in l)
        if (!fpred(x, data)) return false;
    return true;
}

//
// delete_from(cls): delete all instances of class 'cls'
//

function delete_from(|cls) {
    for (x in (eval "select " + cls))
        delete x;
}

//
// delete_(coll): delete contents of collection coll
//

function delete_(coll) {
    for (x in coll)
        delete x;
}

//
// get_from(cls): returns all instances of class 'cls'
//

function get_from(|cls) {
    eval "select " + cls;
}

//
// generates an unused global symbol
//

function gensym() {
    prefix := "::oql#_#_#";
    for (i := 0; ; i++) {
        varname := prefix + string(i);

```

```

    if (!(eval "isset " + varname)) {
        eval varname + " := 0";
        return ident(varname);
    }
}

//
// expression-like for-each function
//

function foreach_expr(|x, |coll, |expr, colltyp ? "list") {
    varname := "_#_#_R_#_#_";

    statement := "push " + varname + " := " + colltyp + "()"; " +
        "for (" + x + " in " + coll + ") " +
        "{" + varname + " += " + expr + "};" +
        "pop " + varname;

    return eval statement;
}

//
// expression-like for-C function
//

function for_expr(|start, |cond, |end, |expr, colltyp ? "list") {
    varname := "_#_#_R_#_#_";

    statement := "push " + varname + " := " + colltyp + "()"; " +
        "for (" + start + "; " + cond + "; " + end + ")" +
        "{" + varname + " += " + expr + "};" +
        "pop " + varname;

    return eval statement;
}

//
// expression-like while-C function
//

function while_expr(|cond, |expr, colltyp ? "list") {
    varname := "_#_#_R_#_#_";

    statement := "push " + varname + " := " + colltyp + "()"; " +
        "while (" + cond + ")" +
        "{" + varname + " += " + expr + "};" +
        "pop " + varname;

    return eval statement;
}

//
// expression-like do/while-C function
//

function do_while_expr(|expr, |cond, colltyp ? "list") {
    varname := "_#_#_R_#_#_";

    statement := "push " + varname + " := " + colltyp + "()"; " +
        "do {" + varname + " += " + expr + "};" +
        "while (" + cond + ");" +
        "pop " + varname;

```

```

    return eval statement;
}

function extentof(|classname) {
    return (select one class.name = classname).extent;
}

function countof(|classname) {
    return (select one class.name = classname).extent.count;
}

function objectcount(db := oql$db) {
    objcnt := 0;
    db->transactionBegin();
    for (cl in (select <db> x from class x where
        x.type != "system" and x.name !~ "<"))
        objcnt += cl.extent.count;
    db->transactionCommit();
    return objcnt;
}

function ifempty(x, y) {
    if (is_empty(x))
        return y;
    return x;
}

function null_ifempty(x) {
    return ifempty(x, null);
}

function getone(x) {
    if (is_empty(x))
        return null;
    return first(flatten(x));
}

//
// database and transaction management
//

function open_db(db_name_or_id, strmode, user := null, passwd := null) {

    if (strmode == "r")
        mode := DBREAD;
    else if (strmode == "rw")
        mode := DBRW;
    else if (strmode == "rlocal")
        mode := DBREAD|DBOPENLOCAL;
    else if (strmode == "rwlocal")
        mode := DBRW|DBOPENLOCAL;
    else
        throw "invalid open mode: r, rw, rlocal or rwlocal expected, got " +
            strmode;

    if (is_int(db_name_or_id))
        db := new<> database(dbid : db_name_or_id);
    else
        db := new<> database(dbname : db_name_or_id);

    if (user == null)
        db.open(oql$db.getConnection(), mode);

```

```

else
    db.open(oql$db.getConnection(), mode, user, passwd);

return db;
}

function set_default(db) {
    db->setDefaultDatabase();
}

function begin(db := oql$db) {
    db->transactionBegin();
}

function begin_params(trsmode, lockmode, recovmode, magorder, ratioalrt, wait_timeout, db := oql$db) {
    db->transactionBegin(trsmode, lockmode, recovmode, magorder, ratioalrt, wait_timeout);
}

function commit(db := oql$db) {
    db->transactionCommit();
}

function abort(db := oql$db) {
    db->transactionAbort();
}

//
// miscellaneous
//

function print_function(f) {
    print "function " + (bodyof f) + "\n";
}

function print_functions() {
    cnt := 0;
    for (f in oql$functions) {
        if (cnt > 0) print "\n";
        print_function(f);
        cnt++;
    }
}

function print_variable(v) {
    print string(v) + " = " + string(eval string(v)) + "\n";
}

function print_variables() {
    for (v in oql$variables) {
        print_variable(v);
        cnt++;
    }
}

function print_classes(system := false) {
    if (system)
        l := (select list(x, x.name) from class x order by x.name);
    else
        l := (select list(x, x.name) from class x where x.type = "user" and x.name !~ "<" order by x.name);

    for (c in l) {
        cls := c[0];
        clsname := c[1];
    }
}

```

```

    print "class " + clsname;
    if (cls.parent != NULL && (system || cls.parent.type != "system"))
        print " extends " + cls.parent.name;
    print "\n";
}
}

function print_obj(o, flags := 0) {
    print o->toString(flags);
}

function print_objs(l, flags := 0) {
    for (o in l)
        print_obj(o, flags);
}

//
// contents_ expression
//

function contents_(coll) {
    r := list();
    for (x in coll) {
        for (s in contents(x))
            r += s;
    }

    return r;
};

function println(s) {
    print(s+"\n");
}

function bench(cmd) {
    t0 := time_stamp::local_time_stamp();
    r := eval cmd;
    t1 := time_stamp::local_time_stamp();
    us := t1->subtract(t0).usecs;
    println("Elapsed time: " + string(us/1000.) + " ms");
    return r;
}

;

```



## 6.5 OQL Quick Reference Card

The following table presents all the OQL statements, expression types and the operators. For the operators common to C++ and OQL, the precedence and associativity is the same.

Quick Reference Card		
Statements		
expression statement	<i>expr</i> ;	
selection statement	if ( <i>cond_expr</i> ) <i>statement</i> [else <i>statement</i> ]	
jump statements	break [ <i>expr</i> ] ; return [ <i>expr</i> ] ;	
iteration statements	while ( <i>cond_expr</i> ) <i>statement</i> do <i>statement</i> while ( <i>cond_expr</i> ) for ( [ <i>expr</i> ] ; [ <i>cond_expr</i> ] ; [ <i>expr</i> ] ) <i>statement</i> for ( <i>var in expr</i> ) <i>statement</i>	
compound statement	{ <i>statement</i> }	
function definition statement	function <i>identifier</i> ([ <i>arglist</i> ]) <i>compound_statement</i>	
empty statement	;	
Arithmetic Expressions		
add	+	<i>expr</i> + <i>expr</i>
subtract	-	<i>expr</i> - <i>expr</i>
multiply	*	<i>expr</i> * <i>expr</i>
divide	/	<i>expr</i> / <i>expr</i>
shift left	<<	<i>expr</i> << <i>expr</i>
shift right	>>	<i>expr</i> >> <i>expr</i>
modular	%	<i>expr</i> % <i>expr</i>
bitwise and	&	<i>expr</i> & <i>expr</i>
bitwise inclusive or		<i>expr</i>   <i>expr</i>
bitwise xor	^	<i>expr</i> ^ <i>expr</i>
complement	~	~ <i>expr</i>
Assignment Expressions		
simple assignment	:=	<i>lvalue</i> := <i>expr</i>
add and assign	+=	<i>lvalue</i> += <i>expr</i>
subtract and assign	-=	<i>lvalue</i> -= <i>expr</i>
multiply and assign	*=	<i>lvalue</i> *= <i>expr</i>
divide and assign	/=	<i>lvalue</i> /= <i>expr</i>
shift left and assign	<<=	<i>lvalue</i> <<= <i>expr</i>
shift right and assign	>>=	<i>lvalue</i> >>= <i>expr</i>
inclusive OR and assign	=	<i>lvalue</i>  = <i>expr</i>
exclusive OR and assign	&=	<i>lvalue</i> &= <i>expr</i>
modulo and assign	%=	<i>lvalue</i> %= <i>expr</i>
exclusive OR and assign	^=	<i>lvalue</i> ^= <i>expr</i>
Auto Increment & Decrement Expressions		
post increment	++	<i>lvalue</i> ++
post decrement	--	<i>lvalue</i> --
pre increment	++	++ <i>lvalue</i>
pre decrement	--	-- <i>lvalue</i>
Logical Expressions		
logical and	&&	<i>expr</i> && <i>expr</i>
logical and	and	<i>expr</i> and <i>expr</i>
logical or		<i>expr</i>    <i>expr</i>
logical or	or	<i>expr</i> or <i>expr</i>
Comparison Expressions		
not	!	! <i>expr</i>
not	not	not <i>expr</i>
equal	=	<i>expr</i> = <i>expr</i>
equal	==	<i>expr</i> == <i>expr</i>
not equal	!=	<i>expr</i> != <i>expr</i>
less than	<	<i>expr</i> < <i>expr</i>
less	<=	<i>expr</i> <= <i>expr</i>
greater	>	<i>expr</i> > <i>expr</i>

greater than	<code>&gt;=</code>	<i>expr &gt;= expr</i>
match regular expression	<code>~</code>	<i>expr ~ expr</i>
match regular expression case insensitive	<code>~~</code>	<i>expr ~~ expr</i>
not match regular expression	<code>!~</code>	<i>expr !~ expr</i>
not match regular expression case insensitive	<code>!~~</code>	<i>expr !~~ expr</i>
match regular expression	<code>like</code>	<i>expr like expr</i>
Conditionnal Expressions		
conditionnal expression	<code>? :</code>	<i>expr ? expr : expr</i>
Expression Lists		
comma sequencing	<code>,</code>	<i>expr , expr</i>
Array Expressions		
subscripting	<code>[]</code>	<i>expr [expr]</i>
interval subscripting	<code>[:]</code>	<i>expr [expr : expr]</i>
Path Expressions		
member selection	<code>.</code>	<i>expr . expr</i>
member selection	<code>-&gt;</code>	<i>expr -&gt; expr</i>
Function Call		
function call	<code>()</code>	<i>expr (expr_list)</i>
Method Invocation		
member selection	<code>()</code>	<i>expr -&gt;expr (arglist)</i>
Eval/Unval Operators		
eval	<code>eval</code>	<i>eval expr</i>
no eval	<code>unval</code>	<i>unval expr</i>
Identifier Expressions		
scope	<code>::</code>	<i>:: identifier</i>
is set	<code>isset</code>	<i>isset identifier</i>
unset	<code>unset</code>	<i>unset identifier</i>
reference of	<code>&amp;</code>	<i>&amp; identifier</i>
	<code>refof</code>	<i>refof identifier</i>
value of	<code>*</code>	<i>* identifier</i>
value of	<code>valof</code>	<i>valof identifier</i>
scope of	<code>scopeof</code>	<i>scopeof identifier</i>
push onto symbol table	<code>push</code>	<i>push identifier</i>
push onto symbol table and assign	<code>push</code>	<i>push expr</i>
pop from symbol table	<code>pop</code>	<i>pop identifier</i>
Set Expressions		
union	<code>union</code>	<i>expr union expr</i>
intersection	<code>intersect</code>	<i>expr intersect expr</i>
except	<code>except</code>	<i>expr except expr</i>
include	<code>&lt;</code>	<i>expr &lt; expr</i>
include or equal	<code>&lt;=</code>	<i>expr &lt;= expr</i>
contain	<code>&gt;</code>	<i>expr &gt; expr</i>
contain or equal	<code>&gt;=</code>	<i>expr &gt;= expr</i>
Object Creation		
new	<code>new</code>	<i>[new] new_construct</i>
new	<code>new</code>	<i>new&lt; opt_expr &gt; new_construct</i>
Object Deletion		
delete	<code>delete</code>	<i>delete expr</i>
Collection Expressions		
contents	<code>contents</code>	<i>contents expr</i>
is in	<code>in</code>	<i>expr in expr</i>
add to collection	<code>add to</code>	<i>add expr to expr</i>
suppress from collection	<code>suppress from</code>	<i>suppress expr from expr</i>
set element in or get element from an indexed collection	<code>[]</code>	<i>expr [ expr ]</i>
append to an indexed collection	<code>append/to</code>	<i>append expr to expr</i>
empty collection	<code>empty</code>	<i>empty expr</i>
exists in collection	<code>in</code>	<i>exists identifier in expr : expr</i>
for all in collection	<code>for all</code>	<i>for all identifier in expr : expr</i>

for some in collection	<b>for</b>	<b>for</b> < <i>expr</i> , <i>expr</i> > <b>in</b> <i>expr</i> : <i>expr</i>
Exception Expressions		
throw exception	<b>throw</b>	<b>throw</b> <i>expr</i>
Function Definition		
define function	<b>define as</b>	<b>define</b> <i>identifier</i> [ <i>arglist as</i> <i>expr</i>
Conversion		
string conversion	<b>string</b>	<b>string</b> ( <i>expr</i> )
integer conversion	<b>int</b>	<b>int</b> ( <i>expr</i> )
character conversion	<b>char</b>	<b>char</b> ( <i>expr</i> )
float conversion	<b>float</b>	<b>float</b> ( <i>expr</i> )
identifier conversion	<b>ident</b>	<b>ident</b> ( <i>expr</i> )
oid conversion	<b>oid</b>	<b>oid</b> ( <i>expr</i> )
Query Expressions		
database query	<b>select</b>	<b>select</b> <i>expr</i> [ <b>from</b> { <i>expr</i> [ <b>as</b> <i>identifier</i> }] [ <b>where</b> <i>expr</i> ]] [ <b>order by</b> { <i>expr</i> }] <b>select</b> <i>expr</i> [ <b>from</b> { <i>identifier in</i> <i>expr</i> } [ <b>where</b> <i>expr</i> ]] [ <b>order by</b> { <i>expr</i> }]
Type Information Expressions		
class of	<b>classof</b>	<b>classof</b> <i>expr</i>
typeof of	<b>typeof</b>	<b>typeof</b> <i>expr</i>
Miscellenaous Expressions		
structure of	<b>structof</b>	<b>structof</b> <i>expr</i>
body of	<b>bodyof</b>	<b>bodyof</b> <i>expr</i>
length pf	[!]	<i>expr</i> [!]
import package	<b>import</b>	<b>import</b> <i>expr</i>